
บทที่   Å
กรณีศึกษาระบบงานขาย และสินคาคงคลัง

จดุประสงคเชิงพฤติกรรม
1. เขาใจระบบขาย และสินคาคงคลังในมุมมองของ Programmer มากขึ้น
2. เขาใจการสรางระบบงานคอมพิวเตอร โดยพิจารณาจากความตองการของผูใช
3. มปีระสบการณ และมองเห็นแนวการออกแบบเมนูทั้งระบบงาน
4. เขาใจแนวการเขียน และการแปลโปรแกรมในอีกรูปแบบหนึ่ง
5. เขาใจการท ํารายงาน และการท ําซํ้ าจากกรณีศึกษานี้มากขึ้น

หวัขอในบทเรียน
6.1  ศึกษาความตองการ
6.2  โครงสรางขอมูล
6.3  วิเคราะหระบบงานเบื้องตน
6.4  โปรแกรมในระบบงานขาย และสินคาคงคลัง

6.4.1  โปรแกรมตาง ๆ ใน MENU.PRG   : สารบัญของโปรแกรม
6.4.2  โปรแกรมตาง ๆ ใน SALE1.PRG  : รวมการขายหลายรูปแบบ
6.4.3  โปรแกรมตาง ๆ ใน SALE2.PRG  : รวมโปรแกรมดานการปรับปรุงขอมูล
6.4.4  โปรแกรมตาง ๆ ใน SALE3.PRG  : รวมโปรแกรมรายงานยอดขาย
6.4.5  โปรแกรมตาง ๆ ใน SALE4.PRG  : รวมโปรแกรมดานสินคาคงคลัง
6.4.6  โปรแกรมตาง ๆ ใน SALE5.PRG  : รวมรายงานอื่น ๆ ที่เกิดขึ้น

ตวัอยางโปรแกรมเกี่ยวกับระบบงานขาย และสินคาคงคลัง  37  ตัวอยาง

หมายเหตุ
ตัวอยางนี้อาจขัดกับความเปนจริงอยูมาก เพราะโปรแกรมขาย หรือสินคาคงคลังนั้น จะมีความซับซอน

อยางมาก ถงึมากทีสุ่ด  กรณีศึกษานี้เกิดจากประสบการณในการเปนลูกคาเปนหลักใหญ  ซึ่งตองการสื่อใหเห็นแนว
การเขยีนโปรแกรมดานการปรับปรุง การทํ ารายงานใหมีความหลากหลาย และซับซอนที่แตกตางกันไป .. หากจะนํ า
ไปใชจริง ตองศึกษาความตองการของผูที่จะนํ าไปใช และพัฒนาตออีกมาก และโปรแกรมทั้งหมดในหนังสือ นัก
ศกึษาสามารถ Download ไปศกึษา ประกอบการเรียนภาคปฏิบัติ

บทที่ 6   กรณีศึกษาระบบงานขาย และสินคาคงคลัง
เนือ่งจากสงัคมทกุวันนี ้ เปนสังคมของนายทุน มีการซื้อขายแลกเปลี่ยนในสังคมตลอดเวลา  ธุรกิจที่หวังผล

ก ําไรจงึอยูไดดวยการขาย และการจัดเก็บสินคาไวในคลัง เพื่อใหการขายไดเงินมาอยางไมตกหลน และการจัดเก็บ
สนิคามปีระสทิธภิาพไมมากไมนอยเกินควร นักการคายุคใหมจึงนํ าคอมพิวเตอรมาใชในหนวยงานกันทั่วหนา

โปรแกรมคอมพิวเตอรที่มีขายอยูในทองตลาด ถูกเขียนขึ้นใหคลอบคลุมกับงานไดในหลาย ๆ ลักษณะมี
ความยดืหยุนสงู  แตมีบางบริษัทที่มีรูปแบบ และความตองการเฉพาะของตน จึงไมอาจนํ าโปรแกรมที่มีขายในทอง
ตลาดมาใชไดอยางเต็มที ่   บริษัทเหลานั้นมักจางนักเขียนโปรแกรม มาพัฒนาระบบคอมพิวเตอรใหตน ตามความ
ตองการที่ไมจ ํากัด

&  6.1 ศึกษาความตองการ
ผูเขยีนไมไดเปนพอคาโดยอาชีพ  จึงศึกษาระบบนี้โดยอาศัยประสบการณตรงของการเปนผูซื้อมาพัฒนา

ระบบ โดยมองตามความนาจะเปน ในเรื่องของงานขาย การจัดทํ ารายงานตลอดจนการควบคุมสินคาคงคลัง และจะ
เนนการขายสง  โดยไมมีการขายหนาราน นั่นคือรับค ําสั่งซื้อทางโทรศัพท จดหมาย หรือโทรสาร เพื่อใหเขาใจระบบ
ไดอยางงาย ๆ

6.1.1 บุคคลที่เกี่ยวของกับงาน
ลํ าดับ ผูเกี่ยวของ ความสัมพันธตอขอมูล

1 ลูกคา สัง่ซื้ออยูที่บาน หรือที่ท ํางาน
2 ลกูจาง หรือพนักงานขาย ปอนขอมูลเขาเครื่องเมื่อมีการสั่งซื้อ หรือฝายอื่น ๆ
3 เจาของกิจการ รบัรายงานการขาย ตาง ๆ
4 ผูผลิตสินคา สงสินคาเขาราน
5 บริษัทรับสงของ เพราะบรษิัทนี้ไมสงของเอง แตจางบริษัทรับสงของมาชวยรับภาระ

 6.1.2 ระบบงานที่เปนอยู และสิ่งที่ผูใชตองการ
ผูเขยีนตองการใหระบบนี ้เปนตัวอยางการพัฒนาระบบที่ทํ าใหผูอานเขาใจไดงาย ๆ จึงควบคุมขอบเขตของ

งานใหอยูในเรื่องการขายสง โดยซื้อสินคาจากผูผลิต รอรับใบสั่งซื้อ  แลวสงสินคาใหลูกคาโดย จางบริษัทขนสงมารับ
ผดิชอบ และระบบนี้จะไมมองถึงหนี้สิน  เพราะความคิดที่วาลูกคาทุกคนเปนคนดี เมื่อรับของแลวจะสงเงินเขาบัญชี
ของบรษิัททันท ีและไมมีปญหาการสงคืนสินคา หรือเช็คเดงแนนอน

ผูเขยีนไดจ ําลองงานตาง ๆ ที่ระบบนี้ควรม ีเปนขอ ๆ ดวยภาษาของผูใชที่ไมมีความรูในคอมพิวเตอรและ
การเขยีนโปรแกรมมากเทาใด จึงไดรายการของความตองการดังตอไปนี้  13

13  ผูเขียนเคยอานหนังสือ 3 เลม คอื การประยุกตคอมพิวเตอรดานการเงิน และการบัญช ีประกอบดวย
ระบบบญัชสีินคาคงคลัง ระบบบัญชีเงินเดือน และระบบบัญชีแยกประเภท ของบริษัท คอมพิวเทคไมโครซิสเท็ม
จํ ากัด พ.ศ.2538 ซึง่มหีลายสวนที่มีอิทธิพลตอการออกแบบระบบกรณีศึกษาในบทนี้

ความตองการของผูใช
1. ขาย ดวยตัวแปรชุดเดียว
2. ขาย ดวยตัวแปรหลายชุด
3. ขาย ดวยตัวแปรอาเรย
4. พมิพรายงานการขาย ตามวันที่
5. พมิพสรุปยอดขายของใบสั่งสินคา ตามวันที่
6. พมิพขอมูลใบสั่งสินคา ตามรหัสพนักงานขาย
7. พมิพรายการสินคาที ่ถึงจุดสั่งซื้อ เรียงตามรหัสสินคา
8. พมิพสรุปยอดขายสินคาแตละชนิด ตามเดือน
9. พมิพสรุปยอดขายสินคาแตละประเภท ตามเดือน
10. ปรบัปรุงแฟมตาง ๆ อยางงาย ๆ
11. พมิพขอมูลจากแฟมอยางงาย ๆ
12. ยกเลิกใบสั่งสินคา
13. ยกเลกิสนิคาในใบสั่งสินคาบางรายการ
14. ปรบัปรุงใบสั่งสินคาใหถูกตอง
15. พมิพขอมูลการขายสินคาในรอบเดือน ตามประเภทสินคา
16. พมิพสรุปคาขนสงใหบริษัทขอสง ตามบริษัทขนสง
17. พมิพสรุปยอดขายสินคา ตามผูผลิตสินคา
18. พมิพสรุปยอดขายสินคา ตามผูผลิตสินคา และเดือน
19. สัง่เพิม่เงนิเดือนพนักงานทุกคน ตามเปอรเซนตที่ตองการ
20. พมิพสถติกิารใชบริการบริษัทขนสงของลูกคาแตละราย ตามชวงวันที่
21. พมิพรายการสินคาที่ขายด ีจากมากไปนอย ตามชวงวันที่
22. พมิพใบสั่งสินคาที ่สงไมทันก ําหนด ตามเดือน
23. พมิพจดหมายขอโทษ ที่สงสินคาไมทันตามกํ าหนด ตามเลขที่ใบสั่งสินคาที่ตองการ
24. พมิพรายการสินคาที่ เหลือไมถึง 10 เปอรเซนต จะถึงจุดสั่งซื้อ
25. พมิพรายชื่อลูกคา ที่พนักงานแตละคนรับผิดชอบ ตามชวงวันที่
26. เพิม่ยอดสินคาเขามาในคลังสินคา
27. รบัสนิคารายการใหมเขามาในคลังสินคา
28. พมิพรายการสินคาที่กํ าลังสั่งซื้อ เพิ่มปจจุบัน
29. พมิพเลขที่ใบสั่งสินคา ที่มีกํ าหนดสงไมเกิน 10 วันจากวันนี้
30. พมิพรายการสินคาที่ถึงจุดสั่งซื้อ ตามผูผลิตสินคา และยังไมไดสั่งซื้อเพิ่ม

&  6.2 โครงสรางขอมูล
โครงสรางขอมูลตอไปนี ้ถูกออกแบบใหรองรับงานตามความตองการขางตน  โดยแยกแฟมออกตาม

ลกัษณะเฉพาะของขอมูล และเชื่อมแฟมเขาดวยกันดวยฟลดหลักหลาย ๆ ฟลดจนเปนฐานขอมูลที่เก็บขอมูลไดไมซํ ้า
ซอน มคีวามกระชับและเรียกขอมูลมาใชรวมกันได
แฟมทีเ่กี่ยวของกับระบบงานขายและสินคาคงคลัง
1. แฟมสั่งสินคา    ORDER.DBF
    แฟมดรรชน ี ORDER1.NTX    จดัเรยีงตาม ORDER,CUST,EMPL

1.1 รหัสใบสั่ง       *ORDER         9(5)
  1.2 รหัสลูกคา       *CUST          9(4)
  1.3 รหัสพนักงาน      *EMPL          9(4)
  1.4 สถานที่จัดสง     SHIPADDR      X(40)
  1.5 วันที่สั่งสินคา     DATEORD       X(8)
  1.6 วนัที่ตองการสินคา  DATEREQ       X(8)
  1.7 วนัที่จัดสงสินคา   DATESHIP      X(8)
  1.8 บริษัทจัดสง      SHIPPER       9(2)
  1.9 คาขนสง          FREIGHT       9(5)V99

  ตวัอยางขอมูล
  95001 1009 3804 444 พระบาท อ.เมือง ล ําปาง 52000  12/09/95 25/09/95 81 18/09/95
  95002 1012 3844 43 บานผา อ.เกาะคา ล ําปาง 52130   19/09/95 26/09/95 83 30/09/95
  95003 1021 3801 21 บานทอง อ.เมือง เลย 44000       21/09/95 28/09/95 81 09/10/95
  95004 1033 3816 5 ต.คลองด ํา อ.เมือง ตาก 63000      25/09/95 06/10/95 82 15/10/95

2. แฟมสั่งสินคาละเอียด ORDERD.DBF
   แฟมดรรชน ี ORDERD1.NTX   จดัเรยีงตาม ORDER,PRO
  ORDERD2.NTX   จดัเรยีงตาม PRO,ORDER
  2.1 รหัสใบสั่ง       *ORDER         9(5)
  2.2 รหัสสินคา       *PRO           9(4)
  2.3 ราคาขาย           OPRICE        9(5)V99
  2.4 ปริมาณขาย         OQUANT        9(4)
  2.5 สวนลด            ODISC         9(4)V99

  ตวัอยางขอมูล
  95001 1105    12.00   20   40.00
  95001 1116  200.50  100   50.00
  95002 1111    81.00   30   30.00
  95002 1121  950.00   10    0.00

3.  แฟมสินคา      PRO.DBF
    แฟมดรรชน ี PRO1.NTX      จดัเรียงตาม PRO
  3.1 รหัสสินคา       *PRO           9(4)
  3.2 รหัสผูผลิด       SUP           9(3)
  3.3 รหัสประเภท        CAT           9(2)
  3.4 ชื่อสินคา        PRONAME       X(40)
  3.5 หนวยที่ใช       PROUNIT       X(10)
  3.6 ราคาทุน           PROPRICE      9(5)V99
  3.7 จ ํานวนเหลือ        PROREST       9(4)
  3.8 จ ํานวนที่สั่งซื้อเพิ่ม  PROONORD 9(4)
  3.9 จดุสั่งซื้อ       REORDER       9(3)

  ตวัอยางขอมูล
  1101 511 14 ปากกา  โหล  25.00   50    0   10
  1102 509 13 กระดาษ รีม  75.00  141  300   150

4. แฟมพนักงาน    EMPL.DBF
    แฟมดรรชน ี EMPL1.NTX     จัดเรียงตาม EMPL
  4.1 รหัสพนักงาน      *EMPL          9(4)
  4.2 ชือ่พนักงาน       EMPLNAME      X(40)
  4.3 ทีอ่ยูพนักงาน    EMPLADDR      X(40)
  4.4 เงินเดือน         EMPLSAL       9(6)V99
  4.5 วนัเกิด           EMPLBORN      X(8)

  ตวัอยางขอมูล
  3801  นาย สมชาย ดีดี  44 ต.สบสงค อ.ดานเกวียน เลย 48000   6500.00  13/06/69
  3802  น.ส.งาน การดี   210 ต.คงคง อ.รักคํ า ล ําพูน 51000       8900.00  21/01/54
  3803  นาย ดง ไผดึก   158 ต.ฐานะ อ.มะระ ล ําปาง 52000       10000.00 09/12/65

5. แฟมลูกคา        CUST.DBF
    แฟมดรรชน ี CUST1.NTX     จดัเรียงตาม CUST
  5.1 รหัสลูกคา      *CUST          9(4)
  5.2 ชือ่ลูกคา        CUSTNAME      X(40)
  5.3 ที่อยูลูกคา     CUSTADDR      X(40)

  ตวัอยางขอมูล
  1001  นาย ไขคํ า ดงดึก    488 ต.ปาดง อ.บานสัน สมุทรสงคราม 16000
  1002  นาย คํ าด ีมีไชย     12 ต.คงคา อ.บานสัน สมุทรสงคราม 16000
  1003  น.ส.เดก็ขาว กากา   1000 ต.ฐานะ อ.มะระ ล ําปาง 52000

6.  แฟมบริษัทรับสงของ   SHIPPER.DBF
     แฟมดรรชน ี SHIPPER1.NTX  จดัเรียงตาม SHIPPER
  6.1 รหัสบริษัทรับสง  *SHIPPER      9(2)
  6.2 ชื่อบริษัทรับสง  SHIPNAME      X(40)

  ตวัอยางขอมูล
  81  แสงตะวันขนสง
  82  แสงหลาขนสง
  83  ค ําดีรักบริการขนสง

7. แฟมประเภทสินคา   CAT.DBF
    แฟมดรรชน ี CAT1.NTX      จดัเรียงตาม CAT
  7.1 รหัสประเภท       *CAT          9(2)
  7.2 ชื่อประเภทสินคา  CATNAME       X(40)

  ตวัอยางขอมูล
  11 เครื่องดื่ม
  12 เครื่องเขียน
  13 ผักและผลไม

8.  แฟมผูผลิต      SUP.DBF
    แฟมดรรชน ี SUP1.NTX      จดัเรียงตาม SUP ตวัอยางขอมูล
  8.1 รหัสผูผลิต      *SUP           9(3) 501   กาแฟเขาชอง จํ ากัด
  8.2 ชื่อผูผลิต       SUPNAME       X(40) 502   ซพีี ประเทศไทย

&   6.3 วิเคราะหระบบงานเบื้องตน
การวเิคราะหระบบงานครั้งนี ้ เนนการจัดทํ าโปรแกรม จึงศึกษาความความตองการ แลวน ําความตองการ

มาจดักลุมเปนระบบเล็ก ๆ   จากนั้นนํ ามาเปนแนวในการจัดทํ าเมนู  เพื่อแบงกลุมของโปรแกรม  ท ําใหมองภาพไดชัด
เจนโดยมคีวามสัมพันธโดยตรงมาจาก สิ่งที่ตองการ

การวเิคราะหระบบงานแบบนี้ไมสามารถน ําไปใชงานจริงได เพราะยังขาดสิ่งสํ าคัญตาง ๆ อีกมาก เชนการ
ศกึษาความเปนไปไดของระบบ  การจัดทํ าผังงาน การศึกษาทิศทางการใชและการเปลี่ยนแปลงของขอมูล เปนตน

6.3.1 จัดระบบยอยจากความตองการของผูใช
ระบบขาย
   1. ขาย ดวยตัวแปรชุดเดียว
   2. ขาย ดวยตัวแปรหลายชุด
   3. ขาย ดวยตัวแปรอาเรย

ระบบปรับปรุง
   1. ปรบัปรุงใบสั่งสินคาใหถูกตอง
   2. ยกเลิกใบสั่งสินคา
   3. ยกเลกิสินคาในใบสั่งสินคาบางรายการ
   4. สัง่เพิม่เงนิเดือนพนักงานทุกคน ตามเปอรเซนตที่ตองการ
   5. ปรบัปรุงแฟมตาง ๆ อยางงาย ๆ
ระบบรายงานยอดขาย
   1. พมิพสรุปยอดขายของใบสั่งสินคา ตามวันที่
   2. พมิพสรุปยอดขายสินคาแตละชนิด ตามเดือน
   3. พมิพสรุปยอดขายสินคาแตละประเภท ตามเดือน
   4. พิมพสรุปยอดขายสินคา ตามผูผลิตสินคา
   5. พมิพสรุปยอดขายสินคา ตามผูผลิดสินคา และเดือน
   6. พมิพรายงานการขาย ตามวันที่
   7. พมิพขอมูลใบสั่งสินคา ตามรหัสพนักงานขาย
   8. พมิพขอมูลการขายสินคาในรอบเดือน ตามประเภทสินคา
ระบบควบคุมสินคาคงคลัง
   1. เพิม่ยอดสินคาเขามาในคลังสินคา
   2. รบัสนิคารายการใหมเขามาในคลังสินคา
   3. พมิพรายการสินคาที่ ถึงจุดสั่งซื้อ เรียงตามรหัสสินคา
   4. พมิพรายการสินคาที่ เหลือไมถึง 10 เปอรเซนต จะถึงจุดสั่งซื้อ
   5. พมิพรายการสินคาที่ก ําลังสั่งซื้อ เพิ่มปจจุบัน
   6. พมิพรายการสินคาที่ถึงจุดสั่งซื้อ ตามผูผลิตสินคา และยังไมไดสั่งซื้อเพิ่ม
   7. พมิพรายการสินคาที่ขายด ีจากมากไปนอย ตามชวงวันที่
ระบบรายงานอื่น
   1. พมิพขอมูลจากแฟมอยางงาย ๆ
   2. พมิพสรุปคาขนสงใหบริษัทขอสง ตามบริษัทขนสง
   3. พมิพสถติกิารใชบริการบริษัทขนสงของลูกคาแตละราย ตามชวงวันที่
   4. พมิพใบสั่งสินคาที ่สงไมทันก ําหนด ตามเดือน
   5. พมิพจดหมายขอโทษ ที่สงสินคาไมทันตามกํ าหนด ตามเลขที่ใบสั่งสินคาที่ตองการ
   6. พมิพรายชื่อลูกคา ที่พนักงานแตละคนรับผิดชอบ ตามชวงวันที่
   7. พมิพเลขที่ใบสั่งสินคา ที่มีกํ าหนดสงไมเกิน 10 วันจากวันนี้

6.3.2 ออกแบบเมนูตามระบบยอย
เมือ่ไดมองเหน็กลุมงานซึ่งถูกแยกเปนระบบยอยแลว   เพื่อใหการออกแบบเมนูงายขึ้น  ในกรณีนี้จึงขอออก

แบบตวัเลอืกตามระบบยอย  แตควรแตงคํ าพูดใหกระชับขึ้น เพื่อจะไดไมยาวจนเกินไปในการทํ าเมนู
เมนูหลัก
1. ระบบขาย
2. ระบบปรับปรุง
3. ระบบรายงานยอดขาย
4. ระบบควบคุมสินคาคงคลัง
5. ระบบรายงานอื่น
6. เลกิการท ํางาน
1. เมนูระบบขาย
   1. ขาย ดวยตัวแปรชุดเดียว
   2. ขาย ดวยตัวแปรหลายชุด
   3. ขาย ดวยตัวแปรอาเรย
   4. กลับเมนูหลัก
2. เมนูระบบปรับปรุง
   1. ปรับปรุงใบสั่งสินคา
   2. ยกเลิกใบสั่งสินคาทั้งใบ
   3. ยกเลิกสินคาในใบสั่งสินคา
   4. เพิม่เงินเดือนพนักงาน
   5. ปรับปรุงแฟมตาง ๆ
   6. กลับเมนูหลัก
3. ระบบรายงานยอดขาย
   1. พมิพสรุปยอดขายของสินคา ตามเดือน
   2. พมิพสรุปยอดขายของสินคา ตามผูผลิตสินคา
   3. พิมพสรุปยอดขายของสินคา ตามผูผลิตสินคา และเดือน
   4. พมิพสรุปยอดขายของสินคาแตละประเภท ตามเดือน
   5. พมิพสรุปยอดขายของใบสั่งสินคา ตามวันที่
   6. พมิพสรุปยอดขายของใบสั่งสินคา ตามรหัสพนักงานขาย
   7. พมิพสรุปยอดขายสินคา ตามประเภทสินคา
   8. พมิพรายการขายสินคาแตละรายการ ตามวันที่
   9. กลับเมนูหลัก

4. ระบบควบคุมสินคาคงคลัง
   1. ซือ้สนิคาเพิ่ม เมื่อสินคาถึงจุดสั่งซื้อ
   2. ซือ้สนิคารายการใหม มาจํ าหนาย
   3. พมิพรายการสินคาที่ ถึงจุดสั่งซื้อ เรียงตามรหัสสินคา
   4. พมิพรายการสินคาที ่ถึงจุดสั่งซื้อ ตามผูผลิต และยังไมไดสั่งเพิ่ม
   5. พมิพรายการสินคาที่ เหลือไมถึง 10 เปอรเซนต จะถึงจุดสั่งซื้อ
   6. พมิพรายการสินคาที ่กํ าลังสั่งซื้อเพิ่มปจจุบัน
   7. พมิพรายการสินคาที ่ขายดี จากมากไปนอย ตามชวงวันที่
   8. กลับเมนูหลัก
5. ระบบรายงานอื่น
   1. พมิพสรุปคาขนสงใหบริษัทขอสง ตามเดือน
   2. พมิพสถติกิารใชบริษัทขนสงของลูกคาแตละราย ตามชวงวันที่
   3. พมิพจดหมายขอโทษ ที่สงสินคาไมทันก ําหนด โดยระบุเลขที่ใบสั่งสินคา
   4. พมิพรายชื่อลูกคา ที่พนักงานแตละคนรับผิดชอบ ตามชวงวันที่
   5. พมิพเลขที่ใบสั่งสินคา ที่สงไมทันก ําหนด ตามเดือน
   6. พมิพเลขที่ใบสั่งสินคา ที่มีกํ าหนดสงไมเกิน 10 วันจากวันนี้
   7. พมิพขอมูลจากแฟมอยางงาย ๆ
   8. กลับเมนูหลัก

6.3.3 หลักการตั้งชื่อโปรแกรม
ส ําหรบัระบบงานขาย และสินคาคงคลัง ไดตั้งชื่อโปรแกรมตามลํ าดับในเมนู  เพื่อสื่อใหเขาใจไดงายโดย

มองล ําดับของตัวเลือกตามเมนู  ส ําหรับชื่อโปรแกรมก ําหนดไว 6 หลัก
หลกัที ่1 ใหเปนตัว S ซึ่งยอมาจาก SALE
หลักที ่2 และ 3 เปนล ําดับในเมนูหลัก
หลกัที ่4 และ 5 เปนล ําดับในเมนูยอย

โปรแกรมตาง ๆ ในระบบทะเบียนมีดังนี้
MENU.PRG : เมนหูลกัส ําหรับระบบงานขาย และสินคาคงคลัง ประกอบดวย 10 โปรแกรมยอย
MENUALL : โปรแกรมหลักที่เรียกเมนูตาง ๆ มาใหเลือก
PRESSDOWN : กระท ําเมื่อกดปุมลูกศรลง เพื่อปรับปรุงคาสูงสุด และตํ ่าสุดของตัวเลือก
PRESSUP : กระท ําเมื่อกดปุมลูกศรขึ้น เพื่อปรับปรุงคาสูงสุด และตํ ่าสุดของตัวเลือก
WRITEMENU : เขยีนเมนูใหมทุกครั้งที่กดลูกศร
ALERTNEW : ใชแทนค ําสั่ง ALERT ท ําใหไดตัวเลือกมากกวา 4 และเปลี่ยนสีได
ASKPRT : สอบถามการปลายทางของผลลัพธ
ASKUPD : ถามยํ ้าการปรับปรุงขอมูล

HELPFORM : แสดงขอความชวยเหลืออยางงาย ๆ เมื่อกดปุมฟงกชัน F1
HELPREADVAR : แสดงคาที่เปนไปไดของฟลดที่ก ําลังรอรับคา
QUITPROC : สัง่เลกิท ํางานของโปรแกรมทั้งหมด
SALE1.PRG : ระบบขาย ประกอบดวย 3 โปรแกรมยอย
S0101 : ขาย ดวยตัวแปรชุดเดียว
S0102 : ขาย ดวยตัวแปรหลายชุด
S0103 : ขาย ดวยตัวแปรอาเรย
SALE2.PRG : ระบบปรับปรุง ประกอบดวย 5 โปรแกรมยอย
S0201 : ปรับปรุงใบสั่งสินคา
S0202 : ยกเลิกใบสั่งสินคาทั้งใบ
S0203 : ยกเลิกสินคาในใบสั่งสินคา
S0204 : เพิม่เงินเดือนพนักงาน
S0205 : ปรับปรุงแฟมตาง ๆ
SALE3.PRG : ระบบรายงานยอดขาย ประกอบดวย 8 โปรแกรมยอย
S0301 : พมิพสรุปยอดขายของสินคา ตามเดือน
S0302 : พมิพสรุปยอดขายของสินคา ตามผูผลิตสินคา
S0303 : พมิพสรุปยอดขายของสินคา ตามผูผลิตสินคา และเดือน
S0304 : พมิพสรุปยอดขายของสินคาแตละประเภท ตามเดือน
S0305 : พมิพสรุปยอดขายของใบสั่งสินคา ตามวันที่
S0306 : พมิพสรปุยอดขายของใบสั่งสินคา ตามรหัสพนักงานขาย
S0307 : พมิพสรุปยอดขายสินคา ตามประเภทสินคา
S0308 : พมิพรายการขายสินคาแตละรายการ ตามวันที่
SALE4.PRG ระบบควบคุมสินคาคงคลัง ประกอบดวย 7 โปรแกรมยอย
S0301 : ซือ้สนิคาเพิ่ม เมื่อสินคาถึงจุดสั่งซื้อ
S0302 : ซือ้สนิคารายการใหม มาจํ าหนาย
S0303 : พมิพรายการสินคาที ่ถึงจุดสั่งซื้อ เรียงตามรหัสสินคา
S0304 : พมิพรายการสินคาที ่ถึงจุดสั่งซื้อ ตามผูผลิต และยังไมไดสั่งเพิ่ม
S0305 : พมิพรายการสินคาที่ เหลือไมถึง 10 เปอรเซนต จะถึงจุดสั่งซื้อ
S0306 : พมิพรายการสินคาที่ กํ าลังสั่งซื้อเพิ่มปจจุบัน
S0307 : พมิพรายการสินคาที ่ขายดี จากมากไปนอย ตามชวงวันที่

SALE5.PRG ระบบรายงานอื่น ประกอบดวย 7 โปรแกรมยอย
S0301 : พมิพสรุปคาขนสงใหบริษัทขอสง ตามเดือน
S0302 : พมิพสถติกิารใชบริษัทขนสงของลูกคาแตละราย ตามชวงวันที่
S0303 : พมิพจดหมายขอโทษ ที่สงสินคาไมทันก ําหนด โดยระบุเลขที่ใบสั่งสินคา
S0304 : พมิพรายชื่อลูกคา ที่พนักงานแตละคนรับผิดชอบ ตามชวงวันที่
S0305 : พมิพเลขที่ใบสั่งสินคา ที่สงไมทันก ําหนด ตามเดือน
S0306 : พมิพเลขที่ใบสั่งสินคา ที่มีกํ าหนดสงไมเกิน 10 วันจากวันนี้
S0307 : พมิพขอมูลจากแฟมอยางงาย ๆ

6.3.4 วิธีการแปลโปรแกรม
การแปลโปรแกรมส ําหรับกรณีศึกษาครั้งนี้  ใชคํ าสั่ง RMAKE ในการแปลโปรแกรม โดยระบุชื่อโปรแกรม

หลัก และโปรแกรมยอยทั้งหมด  ไวในแฟม MENU.RMK
การสรางแฟม MENU.RMK
//  MENU.RMK
//  ========
.PRG.OBJ:

          CLIPPER $< /M /N
.OBJ.EXE:
       SET RTLINKCMD=/POSI
       RTLINK $**, $@ ;
MENU.OBJ:     MENU.PRG
SALE1.OBJ:    SALE1.PRG
SALE2.OBJ:    SALE2.PRG
SALE3.OBJ:    SALE3.PRG
SALE4.OBJ:    SALE4.PRG
SALE5.OBJ:    SALE5.PRG
MENU.EXE: MENU.OBJ  SALE1.OBJ SALE2.OBJ SALE3.OBJ SALE4.OBJ SALE5.OBJ

&  6.4 โปรแกรมในระบบงานขาย และสินคาคงคลัง
โปรแกรมของระบบนี ้ ไดจัดทํ าเปนโปรแกรมยอยไวภายนอก โดยมีโปรแกรม MENU.PRG คอยเรียก

โปรแกรมเล็ก ๆ ที่อยูในโปรแกรมยอยอีกครั้งหนึ่ง  ส ําหรับโปรแกรมยอยภายนอกนั้น แตละโปรแกรมยอยคือระบบ
ยอยระบบหนึง่ เพราะทํ าใหงายตอการปรับปรุงแกไขและคนหา หากประมวลผลตามตัวเลือกแลวผิดพลาด

โปรแกรมยอยจากภายนอกม ี 5 โปรแกรมคือ SALE1.PRG จนถึง SALE5.PRG  รวมโปรแกรมทั้งหมดใน
ระบบนี้เปน 6  โปรแกรม  โดยใหโปรแกรมยอยทั้ง 5 นี้ควบคุมระบบยอย 5 ระบบ

1. ระบบขาย
2. ระบบปรับปรุง
3. ระบบรายงาน
4. ระบบคงคลัง
5. ระบบอื่น

6.4.1 โปรแกรมตาง ๆ ใน MENU.PRG
โปรแกรมใน MENU.PRG นี้มีโปรแกรมยอยทั้งหมด 10 โปรแกรม  โดยรวมโปรแกรมที่เกี่ยวกับเมนู และ

โปรแกรมสนับสนุน เชน โปรแกรมสอบถามปลายทางของผลลัพธ  โปรแกรมถามยํ้ าการปรับปรุงขอมูล  โปรแกรม
แสดงสวนชวยเหลือแบบตาง ๆ   โปรแกรมสั่งเลิกการท ํางาน เปนตน

โปรแกรมเมนูในโปรแกรม MENU.PRG นี ้ ใชหลักการในการตรวจสอบการกดลูกศร ท ําใหเมนูแบบนี้
สามารถแสดงนาฬิกาที่เดิน ขณะรอเลือกเมนูได   เมนูแบบนี้ใหประสิทธิภาพในการใชการ สามารถเพิ่มการทํ างาน
ตาง ๆ  ไดอีกมาก  โดยโปรแกรมเมนูนี้เรียกโปรแกรมยอยจากภายในอีก 3 โปรแกรมคือ PRESSDOWN ,  PRESSUP
และ WRITEMENU   การนํ าโปรแกรมนี้ไปปรับปรุงแกไขและน ําไปใชทํ าไดโดยงาย  เพราะโปรแกรมนี้จะตรวจสอบ
การกดปุมลกูศร แลวท ําการจอภาพใหมทุกครั้ง  แตถากดปุม ENTER จะประมวลผลตามเงื่อนไขที่ระบุตามตัวเลือก
นั้น

ในหวัขอนีจ้ะแสดงตัวอยางโปรแกรมตาง ๆ ที่มีในโปรแกรม MAIN.PRG ทั้งหมด 10 โปรแกรม โดยเริ่มจาก
ตวัอยางที ่6.1 ถงึ ตวัอยางที่ 6.8  (บางโปรแกรมสั้นและมีการท ํางานคลายกัน จึงนํ ามารวมในตัวอยางเดียวกัน)
: ตัวอยางที่ 6.1
&   ผลลัพธ
// โปรแกรมหลัก ของระบบขาย และสินคาคงคลัง เพื่อจัดทํ าเมนูหลักและเมนูยอยใน
// โปรแกรมเดียวกัน เมนูมีลักษณะเปนแบบดึงลง และมีนาฬิกาวิ่งตลอดเวลา
// ซึง่เมนแูบบนี้สามารถปรับปรุง แกไขใหมีประสิทธิภาพไดอีกมากมาย
// และมกีารเรียกโปรแกรมยอยเพื่อจัดท ําเมนูอีก 3 โปรแกรมคือ
// 1. ท ําหนาที่เปลี่ยนคาเมื่อกดลงดวย โปรแกรมยอย PRESSDOWN
// 2. ท ําหนาที่เปลี่ยนคาเมื่อกดขึ้นดวย โปรแกรมยอย PRESSUP
// 3. ท ําหนาที่เขียนเมนูใหมเมื่อเลื่อนตัวเลือกดวย โปรแกรมยอย WRITEMENU

PROCEDURE MENUALL
   SET KEY 28 TO HELPFORM  // ควบคมุปุม F1 เพื่อแสดงขอมูลเกี่ยวกับผูเขียนโปรแกรม
   SET KEY -1 TO HELPREADVAR // ควบคมุปุม F2 เพื่อเลือกขอมูลไดอยางงาย ๆ
   SET KEY -9 TO QUITPROC    // ควบคมุปุม F10 เพื่อถามการเลิกทํ างาน
   SET WRAP ON
   SET DELIMITERS TO '[ ]'
   SET DELIMITERS ON
   SET EPOCH TO 1960
   SET CURSOR OFF
   ARMAIN = 6
   ARCM = ARRAY(ARMAIN)
   ARCM[1] = [1.ระบบขาย]
   ARCM[2] = [2.ระบบปรับปรุง]
   ARCM[3] = [3.ระบบรายงาน]
   ARCM[4] = [4.ระบบคงคลัง]
   ARCM[5] = [5.ระบบอื่น]
   ARCM[6] = [6.เลกิงาน]
   ARS1 = 4
   ARC1 = ARRAY(ARS1)
   ARC1[1] = [1. ขาย ดวยตัวแปรชุดเดียว]
   ARC1[2] = [2. ขาย ดวยตัวแปรหลายชุด]
   ARC1[3] = [3. ขาย ดวยตัวแปรอาเรย]
   ARC1[4] = [4. กลับเมนูหลัก]
   ARS2 = 6
   ARC2 = ARRAY(ARS2)
   ARC2[1] = [1. ปรับปรุงใบสั่งสินคา]
   ARC2[2] = [2. ยกเลกิใบสั่งสินคาทั้งใบ]
   ARC2[3] = [3. ยกเลกิสินคาในใบสั่งสินคา]
   ARC2[4] = [4. เพิม่เงินเดือนพนักงาน]
   ARC2[5] = [5. ปรบัปรุงแฟมตาง ๆ]
   ARC2[6] = [6. กลับเมนูหลัก]
   ARS3 = 9
   ARC3 = ARRAY(ARS3)

   ARC3[1] = [1. พมิพสรุปยอดขายของสินคา ตามเดือน]
   ARC3[2] = [2. พมิพสรุปยอดขายของสินคา ตามผูผลิตสินคา]
   ARC3[3] = [3. พมิพสรุปยอดขายของสินคา ตามผูผลิตสินคา และเดือน]
   ARC3[4] = [4. พมิพสรุปยอดขายของสินคาแตละประเภท ตามเดือน]
   ARC3[5] = [5. พมิพสรุปยอดขายของใบสั่งสินคา ตามวันที่]
   ARC3[6] = [6. พมิพสรปุยอดขายของใบสั่งสินคา ตามรหัสพนักงานขาย]
   ARC3[7] = [7. พมิพสรุปยอดขายสินคา ตามประเภทสินคา]
   ARC3[8] = [8. พมิพรายการขายสินคาแตละรายการ ตามวันที่]
   ARC3[9] = [9. กลับเมนูหลัก]
   ARS4 = 8
   ARC4 = ARRAY(ARS4)
   ARC4[1] = [1. ซือ้สนิคาเพิ่ม เมื่อสินคาถึงจุดสั่งซื้อ]
   ARC4[2] = [2. ซือ้สนิคารายการใหม มาจํ าหนาย]
   ARC4[3] = [3. พมิพรายการสินคาที ่ถึงจุดสั่งซื้อ เรียงตามรหัสสินคา]
   ARC4[4] = [4. พมิพรายการสินคาที ่ถึงจุดสั่งซื้อ ตามผูผลิต และยังไมไดสั่งเพิ่ม]
   ARC4[5] = [5. พมิพรายการสินคาที ่เหลือไมถึง 10 เปอรเซนต จะถึงจุดสั่งซื้อ]
   ARC4[6] = [6. พมิพรายการสินคาที่ กํ าลังสั่งซื้อเพิ่มปจจุบัน]
   ARC4[7] = [7. พมิพรายการสินคาที ่ขายดี จากมากไปนอย ตามชวงวันที่]
   ARC4[8] = [8. กลับเมนูหลัก]
   ARS5 = 8
   ARC5 = ARRAY(ARS5)
   ARC5[1] = [1. พมิพสรุปคาขนสงใหบริษัทขนสง ตามเดือน]
   ARC5[2] = [2. พมิพสถติกิารใชบริษัทขนสงของลูกคาแตละราย ตามชวงวันที่]
   ARC5[3] = [3. พมิพจดหมายขอโทษ ที่สงสินคาไมทันก ําหนด โดยระบุเลขที่ใบสั่งสินคา]
   ARC5[4] = [4. พมิพรายชื่อลูกคา ที่พนักงานแตละคนรับผิดชอบ ตามชวงวันที่]
   ARC5[5] = [5. พมิพเลขที่ใบสั่งสินคา ที่สงไมทันก ําหนด ตามเดือน]
   ARC5[6] = [6. พมิพเลขที่ใบสั่งสินคา ที่มีกํ าหนดสงไมเกิน 10 วันจากวันนี้]
   ARC5[7] = [7. พมิพขอมูลจากแฟมอยางงาย]
   ARC5[8] = [8. กลับเมนูหลัก]
   OPT = 1                                 // ก ําหนดตัวเลือกเริ่มตน
   TOPLMT = OPT                            // ก ําหนดบรรทัดบนสุดของเมนูปจจุบัน
   BOTTOMLMT = ARMAIN                      // ก ําหนดบรรทัดลางสุดของเมนูปจจุบัน
   ALLSUB = ARMAIN                         // จ ํานวนเมนูยอยทั้งหมด

   DO WRITEMENU                            // พิมพเมนูครั้งแรก
   DO WHILE LASTKEY() != 27
      @ 1,5 SAY TIME()                     // แสดงเวลาตลอดเวลา
      INKEY(0.2)                           // หยุดรับแปนพิมพทุก 0.2 วินาที
      IF LASTKEY() != 0
         // กรณกีดปุมใด ๆ จะมาตรวจสอบในสวนนี้
         DO CASE
            CASE LASTKEY() = 27     // กด ESC
                 IF OPT <= ALLSUB
                    DO QUITPROC
                 ELSE
                    OPT = (OPT - MOD(OPT,10))/10
                 ENDIF
            CASE LASTKEY() = 4      // กดขวา
                 IF OPT <= ALLSUB           // ตวัเลือกในเมนูหลัก
                    DO PRESSDOWN
                 ELSE
                    IF (OPT-MOD(OPT,10))/10 = ALLSUB // ขามไปเมนูทางขวาหรือเริ่มใหม
                       OPT = 11
                    ELSE
                       OPT = (OPT-MOD(OPT,10))+11
                    ENDIF
                 ENDIF
            CASE LASTKEY() = 19     // กดซาย
                 IF OPT < 10                    // ตวัเลือกในเมนูหลัก
                    DO PRESSUP
                 ELSE
                    IF (OPT-MOD(OPT,10))/10 = 1 // ขามไปเมนูทางซายหรือสุดทาย
                       OPT = ALLSUB*10+1
                    ELSE
                       OPT = (OPT-MOD(OPT,10))-9
                    ENDIF
                 ENDIF

           CASE LASTKEY() = 24     // กดลง
                DO PRESSDOWN
           CASE LASTKEY() = 5      // กดขึ้น
                DO PRESSUP
           // กรณกีด ENTER จะมาเลือกประมวลผลในสวนนี้
           CASE LASTKEY() = 13     // กดเอ็นเทอร (ENTER)
                DO CASE
                   CASE OPT >= 1 .AND. OPT <= ARMAIN -1
                        OPT = OPT * 10 + 1
                   CASE OPT = ARMAIN
                        @ 20,5 SAY "BYE BYE BYE"
                        DO QUITPROC
                   CASE OPT=10+ARS1 .OR. OPT=20+ARS2 .OR. ;
                        OPT=30+ARS3 .OR. OPT=40+ARS4 .OR. ;
                        OPT=50+ARS5                   // กลับเมนูหลัก
                        OPT = (OPT - MOD(OPT,10))/10
                   OTHERWISE  // กระท ําตัวเลือก
                        ATTAGX = LTRIM(STR(MOD(OPT,10),1))
                        ATSUBX = LTRIM(STR((OPT - MOD(OPT,10))/10,1))
                        SET CURSOR ON
                        CLS
                        DO &("S0" + ATSUBX + "0" + ATTAGX)
                        SET CURSOR OFF
               ENDCASE
           ENDCASE
           // ตัง้คา TOPLMT และ BOTTOMLMT ใหม
           DO CASE
              CASE OPT <10
                 TOPLMT = 1       ; BOTTOMLMT = ARMAIN
              CASE OPT >= 11 .AND. OPT<=10+ARS1
                 TOPLMT = 11      ; BOTTOMLMT = 10+ARS1
              CASE OPT >= 21 .AND. OPT<=20+ARS2
                 TOPLMT = 21      ; BOTTOMLMT = 20+ARS2

PDF created with FinePrint pdfFactory Pro trial version http://www.pdffactory.com

http://www.pdffactory.com


                                                บทที่  6  :  กรณีศึกษาระบบงานขาย และสินคาคงคลัง                                          299

              CASE OPT >= 31 .AND. OPT<=30+ARS3
                 TOPLMT = 31      ; BOTTOMLMT = 30+ARS3
              CASE OPT >= 41 .AND. OPT<=40+ARS4
                 TOPLMT = 41      ; BOTTOMLMT = 40+ARS4
              CASE OPT >= 51 .AND. OPT<=50+ARS5
                 TOPLMT = 51      ; BOTTOMLMT = 50+ARS5
              CASE OPT = 61
                 TOPLMT = 1; BOTTOMLMT = ARMAIN; OPT = 6
           ENDCASE
           DO WRITEMENU           // เขียนเมนูใหม
           KEYBOARD CHR(0)
           INKEY()                // เพือ่ตัง้คาของ LASTKEY() ใหม
       ENDIF
   ENDDO
RETURN
: ตัวอยางที่ 6.2
&   ผลลัพธ
// ใชประกอบโปรแกรมเมนูหลัก โดยกระท ําเมื่อกดปุมลง หรือขึ้น ในการเลือกตัวเลือก
// โดยมเีปาหมายคือปรับปรุงคาของตัวเลือกปจจุบันใหเพื่อขึ้นหรือลดลง
// กระท ําเมื่อกดปุมลงในเมนู
PROCEDURE PRESSDOWN
   IF OPT = BOTTOMLMT // ถาถงึตัวลางสุดแลวยังกดลง
      OPT = TOPLMT   // ให OPT ไปที่ตัวแรก คือ TOPLMT (Top limit)
   ELSE
      OPT = OPT + 1
   ENDIF
RETURN
// กระท ําเมื่อกดปุมขึ้นในเมนู
PROCEDURE PRESSUP
   IF OPT = TOPLMT    // ถาถึงตัวแรกแลวยังกดขึ้น
      OPT = BOTTOMLMT // ให OPT ไปยังตัวสุดทาย
   ELSE
      OPT = OPT - 1
   ENDIF
RETURN

: ตัวอยางที่ 6.3
&   ผลลัพธ
// ใชประกอบโปรแกรมหลัก เพื่อท ําการเขียนจอเมนูใหถูกตองโดยใชคาตัวเลือกปจจุบัน
// การเขยีนเมนูของโปรแกรมนี ้จะเขียนใหมทุกครั้งที่มีการกดปุมใด ๆ
// โปรแกรมยอยนี้จึงใชสํ าหรับเขียนหนาเมนูใหมเมื่อมีการกดแปนพิมพ
PROCEDURE WRITEMENU
   SET COLOR TO W/B
   CLS ; SETPOS(2,0)
   XCOL = ARRAY(ARMAIN)
   FOR J = 1 TO ARMAIN                // พิมพเมนูหลัก
      XCOL[J] = COL() + 1             // เกบ็หลักของเมนูยอยแตละเมนู
      @ 2,XCOL[J] SAY ARCM[J]
   NEXT
   ATTAG = MOD(OPT,10)          // หางของตัวเลือก
   ATSUB = (OPT - ATTAG)/10     // เมนูยอยที่
   IF ATSUB > 0
      I = 4                              // ระบุเลขที่บรรทัดเริ่มตน
      FOR J = I TO I - 1 + &("ARS"+LTRIM(STR(ATSUB,1)))
         @ J,XCOL[ATSUB] SAY &("ARC"+LTRIM(STR(ATSUB,1)))[J-I+1]
      NEXT
      TEMPAR = &("ARC"+LTRIM(STR(ATSUB,1)))
   ENDIF
  SET COLOR TO B/W
  IF OPT > 10                                   // สัง่เลือกตัวเลือกยอย
     @ I-1+ATTAG,XCOL[ATSUB]  SAY TEMPAR[ATTAG] // พมิพตัวเลือกปจจุบัน
     SET COLOR TO W/N
     @ 2,XCOL[ATSUB] SAY ARCM[ATSUB] // พมิพหัวของเมนูยอย
  ELSE
     @ 2,XCOL[ATTAG] SAY ARCM[ATTAG]     // พมิพหัวของเมนูยอยในขณะเลือกหัว
  ENDIF
  SET COLOR TO W/B
RETURN

: ตัวอยางที่ 6.4

&   ผลลัพธ
// ใชแทนฟงกชัน ALERT เพราะทํ าใหใชตัวเลือกไดมากกวา 4 ตัวเลือก และแสดง
// ตวัเลอืกในลักษณะ PULLDOWN CHOICE แลวสามารถระบุสีไดอยางอิสระ
// โดยไมท ําใหสถานะเดิมของสีเปลี่ยนไป เพราะมีการจัดเก็บไวแลว
FUNCTION ALERTNEW
PARAMETER HEADING,CHOICEAR
   CNTAR = 0
   AEVAL(CHOICEAR,{|ELE|CNTAR:=CNTAR+1})
   OLDCOLOR = SETCOLOR()
   SAVE SCREEN TO SCR
   CLS
   SET COLOR TO W/BR
   @ 10,20,12+CNTAR,60 BOX "+-+|+-+| "
   @ 11,21 SAY PADC(HEADING,38)
   FOR I = 1 TO CNTAR
       @ 11+I,30 PROMPT PADR(LTRIM(STR(I))+". "+CHOICEAR[I],30)
   NEXT
   MENU TO VALNEW
   SETCOLOR(OLDCOLOR)
   RESTORE SCREEN FROM SCR
RETURN (VALNEW)
: ตัวอยางที่ 6.5
&   ผลลัพธ
// ตวัอยางนี้ม ี2 ฟงกชันคือ ฟงกชันเพื่อเลือกปลายทางของผลลัพธ
// และฟงกชันถามยํ้ าการปรับปรุงขอมูล เมื่อมีการแกไขขอมูล
// ใหเสียงประกอบการเลือกทุกครั้ง
// ใชถามวาจะพิมพผลลัพธออกทางใด
FUNCTION ASKPRT()
   ASKPRT = ALERTNEW("ผลลพัธของรายงาน",{'จอภาพ','เครื่องพิมพ'})
   IF ASKPRT = 2
      TONE(500,5)
   ELSE
      TONE(800,8)
   ENDIF

RETURN (ASKPRT)
// ใชถามวาจะยอมรับการปรับปรุงขอมูลหรือไม
FUNCTION ASKUPD()
   ASKUPD = ALERTNEW("การปรบัปรุงขอมูล",{'ยอมรับการปรับปรุง','ยกเลิกการปรับปรุง'})
   IF ASKPRT = 2
      TONE(500,5)
   ELSE
      TONE(800,8)
   ENDIF
RETURN (ASKUPD)
: ตัวอยางที่ 6.6
&   ผลลัพธ
// พมิพขอมูลของผูพัฒนาโปรแกรมทางจอภาพ หากกดปุมฟงกชัน F1
// เพราะระบุใน MENUALL วา SET KEY 28 TO HELPFORM
PROCEDURE HELPFORM
   SAVE SCREEN TO SCR
   CLS
   ? "                วนัที่ :",DTOC(DATE())
   ? "               เวลา :",TIME()
   ? "====================================="
   ? "โปรแกรมนี้ถูกพัฒนาโดย. อ.บุรินทร รุจจนพันธุ"
   ? "เริ่มพัฒนาเมื่อ 8 กมุภาพันธ 2545"
   ? "โปรแกรมนี้เปนรุนที ่2.200202"
   ? "หากสนใจ หรือตองการตัวโปรแกรม ติดตอขอรับฟรีไดที"่
   ? "วทิยาลัยโยนก ล ําปาง 52000"
   ? "TEL.(054)265170 EXT.210   FAX.(054)265184"
   INKEY(5)
   RESTORE SCREEN FROM SCR
RETURN

: ตัวอยางที่ 6.7
&   ผลลัพธ
// ฟงกชนันีช้วยใหการใสคาในตัวแปรบางตัวแปรถูกตอง เพราะจะแสดง
// คาทีเ่หมาะสมกับฟลดนั้นมาใหเลือก โดยการอานคาจากแฟมมาแสดงใหดู
// แตจะตองระบุใหชัดเจนวาจะน ําฟลดใดมาใชบาง
// โปรแกรมนี้จึงแสดงขอมูลที่เปนไปไดส ําหรับตัวแปรที่กํ าลังรอรับคา
FUNCTION HELPREADVAR
  OLDALIAS = ALIAS() // เก็บพื้นที่ท ํางานเดิม
  SHOWAR = {}
  DO CASE
     CASE LEFT(READVAR(),4) = "_PRO" // เพราะอาจเปน _PRO[1] ได
          IF OLDALIAS != "PRO"
             USE PRO NEW // ถาเปดแลวตองไมเปดอีก
          ENDIF
          GO TOP
          WHILE !EOF()
              AADD(SHOWAR,STR(PRO)+" | "+PRONAME)
              SKIP
          END
          IND := ACHOICE(2,61,5,78,SHOWAR)
          IF IND > 0 // ปองกันการใชปุม ESC
             IF RIGHT(READVAR(),1) = "]"
                _PRO[J] = VAL(LEFT(SHOWAR[IND],4))
             ELSE
                _PRO = VAL(LEFT(SHOWAR[IND],4))
             ENDIF
          ENDIF
          IF OLDALIAS != "PRO" // เพราะอนุญาติใหเรียก F2 2 ครั้ง
             CLOSE             // ในจุดที่ไมไดเปดแฟมนี้
          ENDIF
     CASE READVAR() = "_ORDER"
          SAVE SCREEN TO SCR
          IF OLDALIAS != "ORDER"
             USE ORDER NEW // ถาเปดแลวตองไมเปดอีก

          ENDIF
          GO TOP
          WHILE !EOF()
              AADD(SHOWAR,STR(ORDER))
              SKIP
          END
          @ 2,69,24,79 BOX "+-+|+-+|"
          IND := ACHOICE(3,70,23,78,SHOWAR)
          IF IND > 0 // ปองกันการใชปุม ESC
             _ORDER = VAL(SHOWAR[IND])
          ENDIF
          IF OLDALIAS != "ORDER" // เพราะอนุญาติใหเรียก F2 2 ครั้ง
             CLOSE             // ในจุดที่ไมไดเปดแฟมนี้
          ENDIF
          RESTORE SCREEN FROM SCR
     OTHERWISE
          X = ALERTNEW("ขอโทษ ไมไดเตรียมสวนสนับสนุน ส ําหรับจุดนี้",{"OK."})
  ENDCASE
  IF LEN(OLDALIAS) > 0
     SELE &OLDALIAS // กลับสูพื้นที่ท ํางานเดิม
  ENDIF
RETURN
: ตัวอยางที่ 6.8
&   ผลลัพธ
// สอบถามการเลิกการท ํางาน หากเลิกการทํ างานจะออกเสียง 3 เสียง
PROCEDURE QUITPROC
   Q =  ALERTNEW("สอบถามการเลิกการท ํางาน",{"เลิกการทํ างาน","ยังไมเลิกงาน"})
   IF Q != 2
      FOR I = 1 TO 3
          TONE(50 * I,5-I)
      NEXT
      QUIT
   ENDIF
RETURN

6.4.2 โปรแกรมตาง ๆ ใน SALE1.PRG
โปรแกรมใน SALE1.PRG นี้มีโปรแกรมยอยภายในทั้งหมด 3 โปรแกรม  โดยรวมโปรแกรมที่เกี่ยวกับระบบ

งานขาย เชน โปรแกรมขายดวยตัวแปรชุดเดียว หลายชุด หรืออาเรย   ทั้ง 3 โปรแกรมนี้เปนโปรแกรมขายทั้งหมด  แต
แตกตางกนัทีห่ลกัการเขียนโปรแกรม เพราะโปรแกรมขายดวยตัวแปรชุดเดียวจะมุงปรับปรุงขอมูลโดยไมมีการ
ค ํานวณยอดตาง ๆ ใหชัดเจน  สวนโปรแกรมขายดวยตัวแปรหลายชุดจะมีตัวแปรมารับผลการคํ านวณตาง ๆ   และ
โปรแกรมขายดวยตัวแปรอาเรยจะท ําการเก็บขอมูลลงอาเรย  สามารถพิมพใบเสร็จโดยอานขอมูลจากตัวแปรอาเรย
มาพิมพได

ระบบงานขายเปนระบบที่ส ําคัญยิ่ง เพราะระบบนี้ถูกพัฒนาขึ้นเพื่อเก็บประวัติการขาย และโปรแกรมใน
ระบบนีจ้ะบนัทึกขอมูลการขายเขาไปเก็บในแฟมสั่งซื้อ ซึ่งเปนขอมูลหลักของระบบ และท ําใหนํ าขอมูลไปปรับปรุง
น ําไปใช และประมวลผลในงานของระบบอื่นไดตอไป

ในหวัขอนีจ้ะแสดงตัวอยางโปรแกรมตาง ๆ ที่มีในโปรแกรม SALE1.PRG ทั้งหมด 3 โปรแกรม โดยเริ่มจาก
ตวัอยางที ่6.9 ถึง ตัวอยางที่ 6.11 ( R0101 - R0103 )
: ตัวอยางที่ 6.9
&   ผลลัพธ
// โปรแกรมรบัขอมูลการขาย โดยใชตัวแปรชุดเดียว คือชุดที่เกี่ยวของกับการปรับปรุง
// หนาที่ของโปรแกรมจะบันทึกขอมูลในแฟมใบสั่งสินคา และใบสั่งสินคาละเอียด
// แลวไปลดยอดสินคาในแฟมสินคา ท ําใหตรวจสอบสินคาคงคลังในภายหลังได
// โดยโปรแกรมจะรับขอมูลทีละรายการ พิมพผลการค ํานวณของรายการ แลวบันทึกขอมูล
PROCEDURE S0101
   _ORDER := _CUST := _EMPL := _SHIPPER := _FREIGHT := 0
   _SHIPADDR = SPACE(40)
   _DATEORD = DATE()
   _DATESHIP = DATE() + 7
   _DATEREQ = CTOD('')
   // ========================
   USE ORDER
   @ 2,5 SAY "ปอนเลขที่ใบสั่งสินคา : " GET _ORDER PICT "99999"
   @ 3,5 SAY "รหัสลูกคา : " GET _CUST PICT "9999"
   @ 4,5 SAY "รหสัพนักงาน : " GET _EMPL PICT "9999"
   @ 5,5 SAY "สถานที่จัดสง : " GET _SHIPADDR
   @ 6,5 SAY "วันที่สั่ง : " GET _DATEORD
   @ 7,5 SAY "วนัที่ตองการ : " GET _DATEREQ
   @ 8,5 SAY "วันที่สง : " GET _DATESHIP

   @ 9,5 SAY "รหัสบริษัทรับขนสง : " GET _SHIPPER PICT "99"
   @ 10,5 SAY "คาขนสง : " GET _FREIGHT PICT "9999.99"
   READ
   APPEND BLANK  // เก็บลงแฟมสั่งสินคา (ORDER.DBF)
   REPLACE ORDER WITH _ORDER,CUST WITH _CUST,EMPL WITH _EMPL,;
           SHIPADDR WITH _SHIPADDR,DATEORD WITH _DATEORD,;
           DATEREQ WITH _DATEREQ,DATESHIP WITH _DATESHIP,;
           SHIPPER WITH _SHIPPER,FREIGHT WITH _FREIGHT
   _SUM = 0
   LN = 11
   DO WHILE LASTKEY() != 27
      USE ORDERD
      _PRO := _OPRICE := _OQUANT := _ODISC := 0
      @ LN,5 SAY "สนิคา" GET _PRO PICT "9999"
      @ LN,18 SAY "ราคาขาย" GET _OPRICE PICT "9999.99"
      @ LN,35 SAY "ปรมิาณ" GET _OQUANT PICT "9999"
      @ LN,52 SAY "สวนลด" GET _ODISC PICT "999.99"
      READ
      IF _PRO != 0
         _SUM = _SUM + (_OPRICE * _OQUANT) - _ODISC
         @ LN, 68 SAY _SUM PICT "9,999,999.99"
         APPEND BLANK    // เกบ็ลงแฟมสั่งสินคาละเอียด (ORDERD.DBF)
         REPLACE ORDER WITH _ORDER,PRO WITH _PRO,OPRICE WITH _OPRICE ,;
            OQUANT WITH _OQUANT,ODISC WITH _ODISC
         // สวนที่ใชลดปริมาณในคลังสินคา
         USE PRO INDEX PRO1
         SEEK _PRO
         IF FOUND()
            REPLACE PROREST WITH PROREST - _OQUANT
         ENDIF
      ELSE
         @ LN+1,68 SAY _SUM PICT "9,999,999.99"
         KEYBOARD CHR(27)

         INKEY()
      ENDIF
      LN++
   ENDDO
   INKEY(3)
RETURN
: ตัวอยางที่ 6.10
&   ผลลัพธ
// โปรแกรมรับขอมูลการขาย โดยใชตัวแปรหลายชุด เพื่อคํ านวณในหลายหนาที่
// หนาที่หลักของโปรแกรมคือ บันทึกขอมูลในแฟมใบสั่งสินคา และใบสั่งสินคาละเอียด
// แลวไปลดยอดสินคาในแฟมสินคา ท ําใหตรวจสอบสินคาคงคลังในภายหลังได
// โดยโปรแกรมจะรับขอมูลทีละรายการ ตรวจจํ านวนในคลังสินคา แลวปรับปรุงขอมูล
// สดุทายแสดงผลรวมเงินของการขายสินคาแตละรายการ ในบรรทัดลางสุด
// แลวสอบถามความตองการพิมพใบเสร็จ ทางเครื่องพิมพหรือจอภาพ
PROCEDURE S0102
   _ORDER := _CUST := _EMPL := _SHIPPER := _FREIGHT := 0
   _SHIPADDR = SPACE(40)  ;    _DATEORD = DATE()
   _DATESHIP = DATE() + 7  ;
   _DATEREQ = CTOD('')
   // ========================
   USE ORDER
   @  2,5 SAY "เลขที่ใบสั่งสินคา :" GET _ORDER PICT "99999"
   @  3,5 SAY "     รหัสลูกคา :" GET _CUST PICT "9999"
   @  4,5 SAY "   รหสัพนักงาน :" GET _EMPL PICT "9999"
   @  5,5 SAY "   สถานที่จัดสง :" GET _SHIPADDR
   @  6,5 SAY "       วันที่สั่ง :" GET _DATEORD
   @  7,5 SAY "   วนัที่ตองการ :" GET _DATEREQ
   @  8,5 SAY "       วันที่สง :" GET _DATESHIP
   @  9,5 SAY " รหัสบริษัทรับขน :" GET _SHIPPER PICT "99"
   @ 10,5 SAY "      คาขนสง :" GET _FREIGHT PICT "9999.99"
   READ
   APPEND BLANK  // เก็บลงแฟมสั่งสินคา (ORDER.DBF)
   REPLACE ORDER WITH _ORDER,CUST WITH _CUST,EMPL WITH _EMPL,;
           SHIPADDR WITH _SHIPADDR,DATEORD WITH _DATEORD,;

           DATEREQ WITH _DATEREQ,DATESHIP WITH _DATESHIP,;
           SHIPPER WITH _SHIPPER,FREIGHT WITH _FREIGHT
   _SUM = 0
   @ 11,10 SAY "รหัสสินคา           ราคา     ปริมาณ  สวนลด  รวมเงิน"
   LN = 12
   DO WHILE LASTKEY() != 27
      _PRO := _OPRICE := _OQUANT := _ODISC := 0
      USE PRO INDEX PRO1
      WHILE .T. // ปอนรหัสสินคาจนกวาจะพบ และมีพอขายใหลูกคา
         @ LN,10 GET _PRO PICT "9999"
         @ LN,35 GET _OPRICE PICT "9999.99"
         @ LN,45 GET _OQUANT PICT "9999"
         @ LN,55 GET _ODISC PICT "999.99"
         READ
         SEEK _PRO
         IF FOUND()
            @ LN,18 SAY LEFT(PRONAME,15)
            IF PROREST >= _OQUANT
               EXIT
            ELSE
               @ LN,18 SAY "ปริมาณมีไมพอขาย"
            ENDIF
         ELSE
            @ LN,18 SAY "ไมพบสินคา"
         ENDIF
         IF _PRO =0
            EXIT
         ENDIF
      END
      IF _PRO != 0
         _SUM = _SUM + (_OPRICE * _OQUANT) - _ODISC
         USE  ORDERD
            APPEND BLANK    // เกบ็ลงแฟมสั่งสินคาละเอียด (ORDERD.DBF)

            REPLACE ORDER WITH _ORDER,PRO WITH _PRO,OPRICE WITH _OPRICE ,;
              OQUANT WITH _OQUANT,ODISC WITH _ODISC
        USE PRO INDEX PRO1   // สวนที่ใชลดปริมาณในคลังสินคา
            SEEK _PRO
            IF FOUND()
               REPLACE PROREST WITH PROREST - _OQUANT
            ENDIF
         @ LN,65 SAY STR((_OPRICE * _OQUANT) - _ODISC)
      ELSE
         @ LN,1 CLEAR TO LN,78
         @ LN,65 SAY _SUM PICT "9,999,999.99"
         INKEY(3)
         KEYBOARD CHR(27)
         INKEY()
      ENDIF
      LN++
   ENDDO
   // การสั่งพิมพใบเสร็จ
   IF _SUM > 0
      GASKPRT = ASKPRT()
      IF GASKPRT = 2
         SET DEVICE TO PRINTER
      ENDIF
         CLS
         SELE 1 ; USE ORDERD
         SET FILTER TO ORDER = _ORDER
         GO TOP
         @ 2,10 SAY "ใบสัง่ซื้อเลขที"่ + STR(_ORDER)
         @ 3,10 SAY "======================="
         SELE 2 ; USE PRO INDEX PRO1  NEW
         LN = 4 ; _SUM = 0
         DO WHILE !EOF() // แฟมหลักในลูปนี้คือ แฟมใบสั่งซื้อละเอียด
            SELE 1  ; _PRO = PRO

            SELE 2  ; SEEK _PRO // แฟมสินคา
               IF FOUND()
                  @ LN,10 SAY PRO
                  @ LN,20 SAY LEFT(PRONAME,18)
                ENDIF
            SELE 1   // แฟมใบสั่งซื้อละเอียด
               @ LN,36 SAY OPRICE
               @ LN,44 SAY OQUANT
               @ LN,52 SAY ODISC
               @ LN,60 SAY (OPRICE * OQUANT) - ODISC
               _SUM = _SUM + (OPRICE *OQUANT) - ODISC
               LN++
            SKIP
         ENDDO
      @ LN,60 SAY _SUM
      IF GASKPRT = 2
         EJECT
      ENDIF
      SET DEVICE TO SCREEN
      WAIT "กดปุมใด ๆ เพื่อท ํางานตอ"
      CLOSE ALL
   ENDIF
   // สอบถามการปอนใบสั่งซื้อสินคารายการใหม
   ASKLOOP = ALERTNEW("สอบถามการปอนใบสั่งซื้อรายการใหม",;
            {"ปอนใบสั่งซื้อใหม","เลิกงานสั่งซื้อ"})
   IF ASKLOOP = 1
      CLS
      DO S0102
   ENDIF
RETURN
: ตัวอยางที่ 6.11
&   ผลลัพธ
// โปรแกรมรับขอมูลการขาย โดยใชตัวแปรอาเรย
// หนาที่หลักของโปรแกรมคือ บันทึกขอมูลในแฟมใบสั่งสินคา และใบสั่งสินคาละเอียด

// แลวไปลดยอดสินคาในแฟมสินคา ท ําใหตรวจสอบสินคาคงคลังในภายหลังได
// โดยโปรแกรมจะรับขอมูลทีละรายการ ตรวจจํ านวนในคลังสินคา แลวปรับปรุงขอมูล
// สดุทายแสดงผลรวมเงินของการขายสินคาแตละรายการ ในบรรทัดลางสุด
// แลวสอบถามความตองการพิมพใบเสร็จ ทางเครื่องพิมพหรือจอภาพ
// ในการพมิพขอมูลจะอานขอมูลจากอาเรยที่เก็บไวมาพิมพ แทนการอานจากแฟมโดยตรง
PROCEDURE S0103
   ASKLOOP = 1
   DO WHILE ASKLOOP = 1
      CLS
      _ORDER := _CUST := _EMPL := _SHIPPER := _FREIGHT := 0
      _SHIPADDR = SPACE(40)
      _DATEORD = DATE()
      _DATESHIP = DATE() + 7
      _DATEREQ = CTOD('')
      // ========================
      USE ORDER
      @  2,5 SAY "เลขที่ใบสั่งสินคา :" GET _ORDER PICT "99999"
      @  3,5 SAY "     รหัสลูกคา :" GET _CUST PICT "9999"
      @  4,5 SAY "   รหัสพนักงาน :" GET _EMPL PICT "9999"
      @  5,5 SAY "   สถานที่จัดสง :" GET _SHIPADDR
      @  6,5 SAY "       วันที่สั่ง :" GET _DATEORD
      @  7,5 SAY "   วนัที่ตองการ :" GET _DATEREQ
      @  8,5 SAY "       วันที่สง :" GET _DATESHIP
      @  9,5 SAY " รหัสบริษัทรับขน :" GET _SHIPPER PICT "99"
      @ 10,5 SAY "      คาขนสง :" GET _FREIGHT PICT "9999.99"
      READ
      IF UPDATED()
         APPEND BLANK  // เก็บลงแฟมสั่งสินคา (ORDER.DBF)
         REPLACE ORDER WITH _ORDER,CUST WITH _CUST,EMPL WITH _EMPL,;
              SHIPADDR WITH _SHIPADDR,DATEORD WITH _DATEORD,;
              DATEREQ WITH _DATEREQ,DATESHIP WITH _DATESHIP,;
              SHIPPER WITH _SHIPPER,FREIGHT WITH _FREIGHT
      ELSE

         RETURN
      ENDIF
      _SUM = 0
      LN = 12 ; J = 1
      _PRO := ARRAY(15)
      _OPRICE := ARRAY(15)
      _OQUANT := ARRAY(15)
      _ODISC := ARRAY(15)
      _MONEY := ARRAY(15)
      @ 11,10 SAY "รหัสสินคา           ราคา     ปริมาณ  สวนลด  รวมเงิน"
      DO WHILE LASTKEY() != 27
         USE PRO INDEX PRO1
         _PRO[J] = 1
         DO WHILE _PRO[J] != 0
            _PRO[J] := _OPRICE[J] := _OQUANT[J] := _ODISC[J] := 0
            @ LN,10 GET _PRO[J] PICT "9999"
            @ LN,35 GET _OPRICE[J] PICT "9999.99"
            @ LN,45 GET _OQUANT[J] PICT "9999"
            @ LN,55 GET _ODISC[J] PICT "999.99"
            READ
            SEEK _PRO[J]
            IF FOUND()
               @ LN,18 SAY LEFT(PRONAME,15)
               IF PROREST >= _OQUANT[J]  // ปรับยอดในคลังสินคา
                  REPLACE PROREST WITH PROREST - _OQUANT[J]
                  _MONEY[J] = _OPRICE[J]*_OQUANT[J]-_ODISC[J]
                  @ LN,65 SAY STR(_MONEY[J])
                  EXIT
               ENDIF
            ENDIF
         ENDDO
         IF _PRO[J] != 0
            _SUM = _SUM + _MONEY[J]

            USE  ORDERD
            APPEND BLANK    // เกบ็ลงแฟมสั่งสินคาละเอียด (ORDERD.DBF)
            REPLACE ORDER  WITH _ORDER,;
                    PRO    WITH _PRO[J],;
                    OPRICE WITH _OPRICE[J],;
                    OQUANT WITH _OQUANT[J],;
                    ODISC  WITH _ODISC[J]
         ELSE
            @ LN,1 CLEAR TO LN,78
            @ LN,65 SAY _SUM PICT "9,999,999.99"
            INKEY(3) ;  KEYBOARD CHR(27) ;  INKEY()
         ENDIF
         LN++  ; J++
      ENDDO
      // การสั่งพิมพใบเสร็จ
      IF _SUM > 0
         ALLPRO = J - 2
         GASKPRT = ASKPRT()
         IF GASKPRT = 2
            SET DEVICE TO PRINTER
         ENDIF
         CLS
         @ 2,10 SAY "ใบสัง่ซื้อเลขที"่ + STR(_ORDER)
         @ 3,10 SAY "======================="
         USE PRO INDEX PRO1
         LN = 4
         FOR I = 1 TO ALLPRO
             SEEK _PRO[I] // แฟมสินคา
             IF FOUND()
                @ LN,10 SAY PRO
                @ LN,20 SAY LEFT(PRONAME,18)
             ENDIF
             @ LN,36 SAY _OPRICE[I]

             @ LN,44 SAY _OQUANT[I]
             @ LN,52 SAY _ODISC[I]
             @ LN,60 SAY _MONEY[I]
             LN++
          NEXT
          @ LN,60 SAY _SUM
          IF GASKPRT = 2 ; EJECT ; ENDIF
          SET DEVICE TO SCREEN
          WAIT "กดปุมใด ๆ เพื่อท ํางานตอ"
          CLOSE ALL
       ENDIF
       ASKLOOP = ALERTNEW("สอบถามการปอนใบสั่งซื้อรายการใหม",;
               {"ปอนใบสั่งซื้อใหม","เลิกงานสั่งซื้อ"})
   ENDDO
RETURN

6.4.3 โปรแกรมตาง ๆ ใน SALE2.PRG
โปรแกรมใน SALE2.PRG นี้มีโปรแกรมยอยภายในทั้งหมด 5 โปรแกรม  โดยรวมโปรแกรมที่เกี่ยวกับระบบ

ปรบัปรงุ เชน โปรแกรมปรับปรุงใบสั่งสินคา  โปรแกรมเพิ่มเงินเดือนใหพนักงาน โปรแกรมยกเลิกใบสั่งสินคา  หรือ
การปรับปรุงแฟมตาง ๆ เปนตน

ระบบปรบัปรงุเปนระบบที่ชวยแกปญหาตาง ๆ เชนปอนขอมูลการสั่งซื้อผิด หรือลูกคายกเลิกใบสั่งซื้อรวม
ถงึการเพิม่เงนิเดือนใหพนักงาน และสามารถแกไขขอมูลในแฟมทุกแฟมไดอยางงาย ๆ

ในหวัขอนีจ้ะแสดงตัวอยางโปรแกรมตาง ๆ ที่มีในโปรแกรม SALE2.PRG ทั้งหมด 5 โปรแกรม โดยเริ่มจาก
ตวัอยางที ่6.12 ถึง ตัวอยางที่ 6.16 ( R0201 - R0205 )
: ตัวอยางที่ 6.12
&   ผลลัพธ
// ปรบัปรุงใบสั่งสินคา โดยรับรหัสใบสั่ง แลวน ําขอมูลในใบสั่งสินคามาใหแกไข
// เมือ่แกไขขอมูลในใบสงสินคาแลว จะนํ าขอมูลในใบสั่งสินคาละเอียดเก็บลงอาเรย
// แลวน ําขอมูลในอาเรยมาใหปรับปรุง เมื่อปรับปรุงแลวจะเขียนทับลงไปในเรคอรดเดิม
PROCEDURE S0201
   _ORDER := 0
   DO WHILE LASTKEY() != 27
      @ 1,5 SAY "เลขที่ใบสั่งซื้อ : " GET _ORDER PICT "99999"
      READ

      USE ORDER
      LOCATE FOR ORDER = _ORDER
      IF FOUND()
         @ 2,5 SAY "รหัสลูกคา : "  GET CUST
         @ 3,5 SAY "รหัสพนักงาน : " GET EMPL
         @ 4,5 SAY "สถานที่จัดสง : " GET SHIPADDR
         @ 5,5 SAY "วนัที่สั่งสินคา : " GET DATEORD
         @ 6,5 SAY "วนัทีต่องการสินคา : "GET DATEREQ
         @ 7,5 SAY "วนัที่จัดสงสินคา : " GET DATESHIP
         @ 8,5 SAY "บริษัทจัดสง : " GET SHIPPER
         @ 9,5 SAY "คาขนสง : " GET FREIGHT
         READ
         CLOSE // ฟลดจะไดรับการปรับปรุงทันท ีโดยไมตอง REPLACE
         SELE 1 ;  USE ORDERD
            SET FILTER TO ORDER = _ORDER
            COUNT TO CNTALL FOR ORDER = _ORDER
            GO TOP
            _PRO := ARRAY(15)
            _OPRICE := ARRAY(15)
            _OQUANT := ARRAY(15)
            _ODISC := ARRAY(15)
            FOR I := 1 TO CNTALL  // เก็บคาลงอาเรย
               _PRO[I] = PRO
               _OPRICE[I] = OPRICE
               _OQUANT[I] = OQUANT
               _ODISC[I] = ODISC
               SKIP
            NEXT
            FOR J := 1 TO CNTALL // รับคาปรับปรุง
                @ J+10, 10 GET _PRO[J]
                @ J+10, 40 GET _OPRICE[J]
                @ J+10, 50 GET _OQUANT[J]
                @ J+10, 57 GET _ODISC[J]

                SELE 2 ; USE PRO INDEX PRO1
                SEEK _PRO[J]
                IF FOUND()
                   @ J+10,18 SAY LEFT(PRONAME,20)
                   @ J+10,65 SAY _OPRICE[J] * _OQUANT[J] - _ODISC[J]
                ENDIF
            NEXT
            READ
            IF UPDATED()  // ตรวจสอบวาปรับปรุงหรือไม
               SELE 1
               GO TOP
               FOR I := 1 TO CNTALL
                  REPLACE PRO WITH _PRO[I],OPRICE WITH _OPRICE[I],;
                         OQUANT WITH _OQUANT[I],ODISC WITH _ODISC[I]
                  SKIP
               NEXT
            ENDIF
         CLOSE ALL
         EXIT // ถาปรับปรุงเสร็จแลว จะเลิกโปรแกรมนี้ทันที
      ENDIF
   ENDDO
RETURN
: ตัวอยางที่ 6.13
&   ผลลัพธ
// ยกเลกิใบสั่งสินคา โดยรับเลขที่ใบสั่งสินคาที่ตองการลบ แลวคนหาเรคอรดที่ตรง
// กบัในแฟมใบสั่งซื้อ และใบสั่งซื้อละเอียด
// เมือ่พบจะลงเรคอรดจากทั้ง 2 แฟมที่มีเลขที่ใบสั่งสินคาตรงกับที่ตองการ
PROCEDURE S0202
   _ORDER = 0
   USE ORDER
   DO WHILE .T.
      @ 5,5 SAY "เลขทีใ่บสั่งซื้อที่ตองการลบ : " GET _ORDER PICT "99999"
      READ
      LOCATE FOR _ORDER = ORDER

      IF FOUND() // เหน็เพยีงเรคอรดเดียวในตารางของ DBEDIT
         DBEDIT(5,10,23,70,{"ORDER","CUST","EMPL"})
         IF ASKUPD() = 1
            DELETE
            PACK
            USE ORDERD
            DELETE ALL FOR ORDER = _ORDER
            PACK
         ENDIF
         EXIT
      ENDIF
   ENDDO
RETURN
: ตัวอยางที่ 6.14
&   ผลลัพธ
// ยกเลกิสินคาในใบสั่งสินคาบางรายการ โดยรับเลขที่ใบสั่งสินคา
// แลวเลอืกเรคอรดที่เกี่ยวของจากแฟมใบสั่งสินคาละเอียดมาใหปรับปรุงโดยงาย
// โดยการแกไขขอมูลใชฟงกชัน BROWSE หลังจากที่เลือกขอมูลที่ตรงกับเลขที่ใบสั่ง
PROCEDURE S0203
   _ORDER = 0
   USE ORDERD
   DO WHILE .T.
      @ 5,5 SAY "เลขทีใ่บสั่งซื้อที่ตองการลบสินคา : " GET _ORDER PICT "99999"
      READ
      LOCATE FOR _ORDER = ORDER
      IF FOUND()
         SET FILTER TO ORDER = _ORDER
         BROWSE(5,10,20,70)
         PACK
         EXIT
      ENDIF
   ENDDO
RETURN

: ตัวอยางที่ 6.15
&   ผลลัพธ
// ปรบัปรงุเงินเดือนพนักงานทุกคน โดยปอนเปอรเซนตของการปรับปรุง
// แลวจงึใชคํ าสั่งปรับปรุงขอมูลพรอมกันใหพนักงานทุกคน
PROCEDURE S0204
   _ADDING = 0
   @ 5,5 SAY "จ ํานวนเปอรเซนตที่เพิ่มเงินเดือนใหพนักงาน : " GET _ADDING
   READ
   IF ASKUPD() = 1
      USE EMPL
      REPLACE ALL EMPLSAL WITH EMPLSAL + (EMPLSAL * _ADDING / 100)
      CLOSE
   ENDIF
RETURN
: ตัวอยางที่ 6.16
&   ผลลัพธ
// ปรบัปรงุแฟมตาง ๆ อยางงาย โดยเลือกแฟมจากตัวเลือก แลวเปดแฟม
// จากนัน้ใชฟงกชัน BROWSE ชวยใหการปรับปรุงขอมูลงายขึ้นมาก
PROCEDURE S0205
   OMENU = ALERTNEW("เลอืกปรับปรุงแฟมตามตองการ",;
          {"แฟมสั่งสินคา","แฟมสั่งสินคาละเอียด","แฟมสินคา",;
           "แฟมพนักงาน","แฟมลูกคา","แฟมบริษัทรับสงของ",;
           "แฟมประเภทสินคา","แฟมผูผลิต","กลับเมนู"})
   DO CASE
      CASE OMENU = 1 ; USE ORDER
      CASE OMENU = 2 ; USE ORDERD
      CASE OMENU = 3 ; USE PRO
      CASE OMENU = 4 ; USE EMPL
      CASE OMENU = 5 ; USE CUST
      CASE OMENU = 6 ; USE SHIPPER
      CASE OMENU = 7 ; USE CAT
      CASE OMENU = 8 ; USE SUP
   ENDCASE
   IF OMENU >= 1 .AND. OMENU <= 8

      BROWSE(5,5,22,75)
      PACK
   ENDIF
   CLOSE ALL
RETURN

6.4.4 โปรแกรมตาง ๆ ใน SALE3.PRG
โปรแกรมใน SALE3.PRG นี้มีโปรแกรมยอยภายในทั้งหมด 8 โปรแกรม  โดยรวมโปรแกรมที่เกี่ยวกับระบบ

รายงาน เชน โปรแกรมพิมพรายงานยอดขายของสินคา ตามเงื่อนไขตาง ๆ   โปรแกรมพิมพรายงานใบสั่งสินคาตาม
เงือ่นไขตาง ๆ   และโปรแกรมสรุปใบสั่งซื้อของแตละวัน

ระบบรายงานเปนระบบที่ชวยท ําใหเห็นคุณคา และผลงานจากระบบงานขาย  ท ําใหผูบริหารบริหารสินคา
ในคลงั วางแผนงานหรือนโยบายทางการตลาดไดอยางถูกตอง เพราะรายงานจะแสดงยอดขาย แสดงการรับบริการ
ของลูกคา  เหน็ยอดจ ําหนายของสินคาแตละประเภท  จัดการกับสินคาที่ขายไมได หรือสินคาที่ขายดี ใหขายดียิ่ง ๆ
ขึน้  เพื่อใหบริษัทเจริญรุงเรือง และเติบโตตอไป

ในหวัขอนีจ้ะแสดงตัวอยางโปรแกรมตาง ๆ ที่มีในโปรแกรม SALE3.PRG ทั้งหมด 8 โปรแกรม โดยเริ่มจาก
ตวัอยางที ่6.17 ถึง ตัวอยางที่ 6.24 ( R0301 - R0308 )
: ตัวอยางที่ 6.17
&   ผลลัพธ
// พมิพสรุปยอดขายของสินคา โดยระบุเดือนที่สั่งสินคา
// รายงานแสดงรหัสสินคา และจ ํานวนเงิน ที่ไดทั้งหมด ตามเดือน
PROCEDURE S0301
   USE ORDERD
   SORT ON PRO,ORDER TO TEMPF
   USE TEMPF
   COPY TO ORDERD
   _MONTH = 0
   @ 5,5 SAY "เดอืนที่ตองการ : " GET _MONTH PICT "99"
   READ
   SELE 1 ; USE ORDER
   SELE 2 ; USE  ORDERD
   _PROO = 0
   _TQUANT = 0

   CHKLINE = 1
   DO WHILE !EOF()
      SELE 2  // แฟมใบสั่งสินคาละเอียด
         _ORDER = ORDER
      SELE 1  // แฟมใบสั่งสินคา
         _FOUNDM = 0
         LOCATE FOR ORDER = _ORDER
         IF FOUND()
            IF MONTH(DATEORD) = _MONTH  // ตรวจเดือนที่ตองการ
               _FOUNDM = 1
            ENDIF
         ENDIF
      SELE 2 // แฟมใบสั่งสินคา
         IF _FOUNDM = 1
            IF PRO != _PROO // ถาตางกลุม
               ? PRO
               _PROO = PRO
               CHKLINE = 0
            ENDIF
            _TQUANT = _TQUANT + OQUANT
         ENDIF
         SKIP
         IF (_PROO != PRO .OR. EOF()) .AND. CHKLINE = 0
            ?? _TQUANT  // ถาเปลี่ยนกลุม
            _TQUANT = 0
            CHKLINE = 1 // แสดงการพิมพบรรทัดแตละบรรทัดสมบูรณ
         ENDIF
   ENDDO
   CLOSE ALL
   INKEY(3)
RETURN

: ตัวอยางที่ 6.18
&   ผลลัพธ
// พมิพสรุปยอดขายของสินคา โดยระบุผูผลิตสินคา
// รายงานแสดงรหัสสินคา ชื่อสินคา ปริมาณขาย และจ ํานวนเงินของสินคา ตามผูผลิต
PROCEDURE S0302
   _SUP = 0
   @ 5,5 SAY "ผูผลิตสินคา : " GET _SUP PICT "999"
   READ
   SELE 1 ; USE PRO INDEX PRO1
   SELE 2 ; USE ORDERD
   INDEX ON PRO TO ORDERD2
   SET RELATION TO PRO INTO PRO
   _PROO = 0
   _TQUANT = 0
   _TMONEY = 0
   DO WHILE !EOF()
      IF PRO->SUP = _SUP
         IF _PROO != PRO // ถาตางกลุม
            ? PRO,PRO->PRONAME
            _PROO = PRO
         ENDIF
         _TQUANT = _TQUANT + OQUANT
         _TMONEY = _TMONEY + (OPRICE * OQUANT - ODISC)
         SKIP
         IF _PROO != PRO .OR.  EOF()
            ?? _TQUANT,_TMONEY  // ถาเปลี่ยนกลุม
            _TQUANT := _TMONEY := 0
         ENDIF  // ตรวจกลุมของสินคา
      ELSE // ถาไมใชผูผลิตที่ตองการ
         SKIP
      ENDIF // ตรวจผูผลิต
   ENDDO
   CLOSE ALL
   INKEY(3)

RETURN
: ตัวอยางที่ 6.19
&   ผลลัพธ
// พมิพสรุปยอดขายของสินคา โดยระบุผูผลิตสินคา และเดือนที่สั่ง
// รายงานแสดงรหัสสินคา ชื่อสินคา และปริมาณ ตามผูผลิตและเดือน
// และแสดงใหเห็นการเชื่อมแฟมสินคา ใบสั่งสินคา และใบสั่งสินคาละเอียด
PROCEDURE S0303
   _SUP := _MONTH := 0
   @ 5,5 SAY "เดอืนที่ตองการ : " GET _MONTH PICT "99"
   @ 6,5 SAY "ผูผลิตสินคา : " GET _SUP PICT "999"
   READ
   SELE 1 ; USE PRO INDEX PRO1
   SELE 2 ; USE ORDER
   SELE 3 ; USE ORDERD
   INDEX ON PRO TO ORDERD2
   SET RELATION TO PRO INTO PRO
   _PROO = 0
   _TQUANT = 0
   CHKLINE = 1
   DO WHILE !EOF()
      _ORDER = ORDER
      SELE 2  // แฟมใบสั่งสินคา
         _FOUNDM = 0
         LOCATE FOR _ORDER = ORDER .AND. MONTH(DATEORD) = _MONTH
         IF FOUND()
            _FOUNDM = 1
         ENDIF
      SELE 3 // แฟมใบสั่งสินคา
         IF _FOUNDM = 1 .AND. PRO->SUP = _SUP
            IF PRO != _PROO  // ถาตางกลุม
               ? PRO,PRO->PRONAME
               _PROO = PRO
               CHKLINE = 0
            ENDIF

            _TQUANT = _TQUANT + OQUANT
         ENDIF
         SKIP
         IF (_PROO != PRO .OR. EOF()) .AND. CHKLINE = 0
            ?? _TQUANT  // ถาเปลี่ยนกลุม
            _TQUANT = 0
            CHKLINE = 1 // แสดงการพิมพบรรทัดแตละบรรทัดสมบูรณ
         ENDIF
   ENDDO
   CLOSE ALL   ;   INKEY(3)
RETURN
: ตัวอยางที่ 6.20
&   ผลลัพธ
// พมิพสรุปยอดขายของประเภทสินคา โดยระบุเดือนที่สั่ง
// รายงานแสดงรหัสใบสั่ง รหัสสินคา ชื่อสินคา และจ ํานวนเงินในใบสั่ง ตามประเภท
// แลวสรุปยอดขายของสินคาแตละประเภท
// และแสดงใหเห็นการเชื่อมแฟมสินคา ใบสั่งสินคา และใบสั่งสินคาละเอียด
PROCEDURE S0304
   _MONTH := 0
   @ 5,5 SAY "เดอืนที่ตองการ : " GET _MONTH PICT "99"
   READ
   SELE 1 ; USE PRO INDEX PRO1
   SELE 2 ; USE ORDER
   SELE 3 ; USE ORDERD INDEX ORDERD2
   SET RELATION TO PRO INTO PRO
   SELE 4 ; USE CAT
   _CAT = 0
   _TQUANT = 0
   DO WHILE !EOF()
      _CAT = CAT
      ? CAT,CATNAME
      SELE 3 // แฟมใบสั่งสินคาละเอียด
         SET FILTER TO PRO->CAT = _CAT // จับกลุมใหประเภทสินคา
         GO TOP

         _TQUANT = 0
         DO WHILE !EOF()
            _ORDER = ORDER
            SELE 2   // แฟมใบสั่งสินคา
            LOCATE FOR ORDER = _ORDER .AND. MONTH(DATEORD) = _MONTH
            IF FOUND()
                 SELE 3
                 _TQUANT = _TQUANT + OQUANT
                 ? ORDER,PRO,PRO->PRONAME,OQUANT
            ENDIF
            SELE 3
            SKIP
         ENDDO
      SELE 4 // แฟมประเภทสินคา
      ? "         จํ านวน = ",_TQUANT
      ? "==============================="
      SKIP
   ENDDO
   CLOSE ALL
   INKEY(3)
RETURN

: ตัวอยางที่ 6.21
&   ผลลัพธ
// พมิพสรุปยอดขายใบสั่งสินคาแตละใบ โดยระบุวันที่
// รายงานแสดงล ําดับ รหัสใบสั่ง และยอดขายของใบสั่งสินคา ตามวันที่
// โดยการหายอดขายจะใชประโยชนจากตัวแปรอาเรยมาเก็บขอมูล
// แลวจงึน ําขอมูลจากอาเรยมาพิมพอีกครั้งหนึ่ง
PROCEDURE S0305
   SELE 1 ; USE ORDERD
   SELE 2 ; USE ORDER
   DO WHILE LASTKEY() != 27
      _DATEORD = CTOD('')
      @ 5,5 SAY "วนัที่ที่มีการสั่งสินคา :" GET _DATEORD

      READ
      SELE 2
      COUNT TO CNTORDER FOR DATEORD = _DATEORD
      IF CNTORDER > 0
         @ 6,5 SAY "จ ํานวนใบสั่งซื้อ :" + STR(CNTORDER)
         SET FILTER TO DATEORD = _DATEORD
         GO TOP
         ARORDER = {}
         DO WHILE !EOF()
            AADD(ARORDER,ORDER)
            SKIP
         ENDDO
         ARMONEY = ARRAY(CNTORDER)
         AFILL(ARMONEY,0)
         SELE 1  // เริม่หาจ ํานวนเงินของใบสั่งแตละใบ ในแฟมใบสั่งสินคาละเอียด
         FOR I = 1 TO CNTORDER
             SET FILTER TO ORDER = ARORDER[I]
             GO TOP
             _MONEY = 0
             DO WHILE ! EOF()
                _MONEY = _MONEY + OPRICE * OQUANT - ODISC
                SKIP
             ENDDO
             ARMONEY[I] = _MONEY
          NEXT
          @ 8,5 SAY "ล ําดับ รหัสใบสั่งสินคา จํ านวนเงิน"
          FOR I = 1 TO CNTORDER
              @ 8+I,5 SAY STR(I,4)+STR(ARORDER[I],6)+STR(ARMONEY[I],9)
          NEXT
      ELSE
          @ 8,5 SAY "NOT FOUND"
      ENDIF
      INKEY(0)

      @ 8,5 CLEAR TO 20,70
   ENDDO
RETURN
: ตัวอยางที่ 6.22
&   ผลลัพธ
// พมิพสรุปยอดขายใบสั่งสินคาแตละใบ โดยระบุรหัสพนักงานขาย
// รายงานแสดงล ําดับ รหัสใบสั่ง และยอดขายของใบสั่งสินคา ตามรหัสพนักงานขาย
// ในการปอนรหัสพนักงานขาย ตองปอนจนกวาจะมีรหัสนั้นในแฟมจริง ๆ จึงจะยอมรับ
// โดยการหายอดขายจะใชประโยชนจากตัวแปรอาเรยมาเก็บขอมูล
// แลวจงึน ําขอมูลจากอาเรยมาพิมพอีกครั้งหนึ่ง ซึ่งเลือกปลายทางของผลลัพธได
PROCEDURE S0306
   SELE 1 ; USE EMPL
   SELE 2 ; USE ORDERD
   SELE 3 ; USE ORDER
   DO WHILE LASTKEY() != 27
      _EMPL = 0
      SELE 1
         DO WHILE LASTKEY() != 27
            @ 5,5 SAY "รหสัพนักงาน :" GET _EMPL PICT "9999"
            READ
            LOCATE FOR EMPL = _EMPL
            IF FOUND()
               @ 7,5 SAY EMPLNAME
               EXIT
            ENDIF
         ENDDO
      SELE 3
      COUNT TO CNTEMPL FOR EMPL = _EMPL
      IF CNTEMPL > 0
         GASKPRT = ASKPRT()    // สอบถามการนํ าออกเครื่องพิมพ
         IF GASKPRT = 2
            SET DEVICE TO PRINTER
            @ 5,5 SAY "รหสัพนักงาน : "+STR(_EMPL)
         ENDIF

         @ 6,5 SAY "จ ํานวนใบสั่งซื้อ :" + STR(CNTEMPL)
         SET FILTER TO EMPL = _EMPL
         GO TOP
         ARORDER = {}
         DO WHILE !EOF()
            AADD(ARORDER,ORDER)
            SKIP
         ENDDO
         ARMONEY = ARRAY(CNTEMPL)
         AFILL(ARMONEY,0)
         _TMONEY = 0
         SELE 2  // เริม่หาจ ํานวนเงินของใบสั่งแตละใบ ในแฟมใบสั่งสินคาละเอียด
         FOR I = 1 TO CNTEMPL
             SET FILTER TO ORDER = ARORDER[I]
             GO TOP
             _MONEY = 0
             DO WHILE ! EOF()
                _MONEY = _MONEY + OPRICE * OQUANT - ODISC
                SKIP
             ENDDO
             ARMONEY[I] = _MONEY
             _TMONEY = _TMONEY + _MONEY
          NEXT
          @ 8,5 SAY "ล ําดับ รหัสใบสั่งสินคา จํ านวนเงิน"
          FOR I = 1 TO CNTEMPL
              @ 8+I,5 SAY STR(I,4)+STR(ARORDER[I],6)+STR(ARMONEY[I],9)
          NEXT
          @ 9+CNTEMPL,5 SAY "รวมเงิน : "+STR(_TMONEY)
          IF GASKPRT = 2
             EJECT
          ENDIF
          SET DEVICE TO SCREEN
      ELSE

          @ 8,5 SAY "NOT FOUND"
      ENDIF
      INKEY(0)
      @ 7,5 CLEAR TO 20,70
   ENDDO
RETURN
: ตัวอยางที่ 6.23
&   ผลลัพธ
// พมิพสรุปยอดขายสินคา โดยระบุประเภท
// รายงานแสดงยอดขายของสินคาแตละรายการในแฟมใบสั่งสินคาละเอียด ตามประเภท
// โดยแสดงล ําดับ เลขที่ใบสั่งซื้อ ชื่อสินคา และจ ํานวนเงิน บรรทัดสุดทายแสดงยอดรวม
// สามารถเลือกปลายทางของผลลัพธได
PROCEDURE S0307
   _CAT := 0
   @ 5,5 SAY "ประเภทสินคา : " GET _CAT PICT "99"
   READ
   USE PRO INDEX PRO1
   USE ORDERD NEW
   SET RELATION TO PRO INTO PRO
   SET FILTER TO PRO->CAT = _CAT
   COUNT TO CNTCAT
   GO TOP
   IF CNTCAT > 0
      GASKPRT = ASKPRT()
      IF GASKPRT = 2
         SET PRINT ON
      ENDIF
      SETPOS(7,5);  ? _CAT," : ",CNTCAT," RECORDS"
      I = 1  ;      _TMONEY = 0
      DO WHILE !EOF()
         _MONEY = OPRICE * OQUANT - ODISC
         _TMONEY = _TMONEY + _MONEY
         SETPOS(I+7,5); ? I,ORDER,LEFT(PRO->PRONAME,20),_MONEY
         SKIP

         I++
      ENDDO
      SETPOS(I+7,5); ? "เปนเงินทั้งหมด = ",_TMONEY
      IF GASKPRT = 2
         EJECT
         SET PRINT OFF
      ENDIF
   ELSE
      @ 6,5 SAY "NOT FOUND"
   ENDIF
   INKEY(5)
   @ 6,5 CLEAR TO 25,75
   CLOSE ALL
RETURN
: ตัวอยางที่ 6.24
&   ผลลัพธ
// พมิพรายการขายแตละรายการ โดยระบุวันที่
// รายงานแสดงล ําดับ รหัสสินคา ราคา ปริมาณ และสวนลด ตามวันที่
// และไดผลลัพธทางจอภาพเทานั้น
PROCEDURE S0308
   _DATEORD := CTOD('')
   @ 5,5 SAY "วนัที่ลูกคาสั่ง : " GET _DATEORD
   READ
   SELE 1 ; USE ORDERD
   INDEX ON STR(ORDER)+STR(PRO) TO ORDERD1
   SELE 2 ; USE ORDER
      SET FILTER TO  DATEORD = _DATEORD
      COUNT TO CNTORD
   IF CNTORD > 0
      GO TOP ; SETPOS(7,1)
      DO WHILE !EOF()
         _ORDER = ORDER
         SELE 1
         SET FILTER TO ORDER = _ORDER

         ? "*",_ORDER
         GO TOP ; I = 1
         DO WHILE !EOF()
            ? I,PRO,OPRICE,OQUANT,ODISC
            SKIP ; I++
         ENDDO
         ? "==============================="
         SELE 2 ; SKIP
      ENDDO
   ELSE
      @ 8,5 SAY "NOT FOUND"
   ENDIF
   INKEY(3)   ;   CLOSE ALL
RETURN

6.4.5 โปรแกรมตาง ๆ ใน SALE4.PRG
โปรแกรมใน SALE4.PRG นี้มีโปรแกรมยอยภายในทั้งหมด 7 โปรแกรม  โดยรวมโปรแกรมที่เกี่ยวกับระบบ

สนิคาคงคลงั เชน โปรแกรมพิมพรายงานสินคาที่ถึงจุดสั่งซื้อ โปรแกรมเพิ่มจํ านวนสินคา โปรแกรมเพิ่มรายการสินคา
ใหมในคลงั  โปรแกรมรายงานเกี่ยวกับจํ านวนสินคาแบบตาง ๆ

ระบบสนิคาคงคลงัเปนระบบที่ชวยท ําใหบริษัทมีสินคาจํ าหนายตลอดเวลาเมื่อลูกคาถาม และไมมีสินคาใน
คลงัมากเกนิไป เพราะการสั่งสินคามาก ๆ ท ําใหเงินทุนจมอยูในคลัง  ซึ่งเปนการบริหารเงินที่ไมดีนัก  ดังนั้นในระบบ
นีจ้งึชวยผูบรหิารใจการตัดสินใจซื้อสินคาเพิ่มเมื่อใด เทาใด และจากใคร

ในหวัขอนีจ้ะแสดงตัวอยางโปรแกรมตาง ๆ ที่มีในโปรแกรม SALE4.PRG ทั้งหมด 7 โปรแกรม โดยเริ่มจาก
ตวัอยางที ่6.24 ถึง ตัวอยางที่ 6.30 ( R0401 - R0407 )
: ตัวอยางที่ 6.24
&   ผลลัพธ
// เพิม่ยอดสนิคาเขามาในคลังสินคา ของสินคาแตละรายการ
// มกักระท ําเมื่อสินคาในคลังหมด แลวซื้อเพิ่ม จึงตองเพิ่มจํ านวนสินคาในแฟม
// โดยเพิม่ยอดสินคาไดทีละรายการ หลายรายการ จนกวาจะกดปุม ESC
PROCEDURE S0401
   USE PRO INDEX PRO1
   LN = 5
   DO WHILE LASTKEY() != 27
      _PRO := _PROREST := 0
      @ LN,10 SAY "รหัสสินคา : " GET _PRO PICT "9999"   ;     READ
      SEEK _PRO
      IF FOUND()
         @ LN,30 SAY LEFT(PRONAME,15)
         @ LN,50 SAY "เพิม่จ ํานวนสินคาอีก :" GET _PROREST PICT "999"
         READ
         IF ASKUPD() = 1
            REPLACE PROREST WITH PROREST + _PROREST,PROONORD WITH 0
         ENDIF
         LN++
      ELSE
         @ LN,50 SAY "NOT FOUND"
      ENDIF
   ENDDO
RETURN

: ตัวอยางที่ 6.25
&   ผลลัพธ
// เพิม่ขอมลูสนิคาใหมเขาไปในแฟมสินคา เมื่อมีสินคาชนิดใหมที่ไมเคยมีจํ าหนาย
// เมือ่รอรับขอมูลจะแสดงรหัสสินคารายการใหมมาใหอัตโนมัติ
// โดยเพิม่รายการสินคาไดทีละรายการ หลายรายการจนกวาจะกดปุม ESC
PROCEDURE S0402
   USE PRO INDEX PRO1
   _PRO = 1
   DO WHILE _PRO != 0 .AND. LASTKEY() !=27
      GO BOTTOM
      _PRO = PRO + 1
      _SUP = 0   ;      _CAT = 0
      _PRONAME = SPACE(40)
      _PROUNIT = SPACE(10)
      _PROPRICE := _PROREST := _PROONORD := _REORDER := 0
      @ 10,5 SAY "รหสัสินคา" GET _PRO PICT "9999"
      @ 11,5 SAY "รหัสผูผลิต" GET _SUP PICT "999"
      @ 12,5 SAY "รหัสประเภท" GET _CAT PICT "99"
      @ 13,5 SAY "ชือ่สินคา" GET _PRONAME
      @ 14,5 SAY "ชือ่หนวย" GET _PROUNIT
      @ 15,5 SAY "ราคาทุน" GET _PROPRICE PICT "9999.99"
      @ 16,5 SAY "จ ํานวนที่เหลือ" GET _PROREST PICT "9999"
      @ 17,5 SAY "จ ํานวนที่ก ําลังสั่งเพิ่ม" GET _PROONORD PICT "9999"
      @ 18,5 SAY "จดุสัง่ซื้อ" GET _REORDER PICT "999"
      READ
      IF UPDATED() .AND. ASKUPD() = 1
         APPEND BLANK
         REPLACE PRO WITH _PRO,SUP WITH _SUP,CAT WITH _CAT,;
                 PRONAME WITH _PRONAME,PROUNIT WITH _PROUNIT,;
                 PROPRICE WITH _PROPRICE,PROREST WITH _PROREST,;
                 PROONORD WITH _PROONORD
      ENDIF
   ENDDO
RETURN

: ตัวอยางที่ 6.26
&   ผลลัพธ
// พมิพรายการสินคาที ่ถึงจุดสั่งซื้อ เรียงตามรหัสสินคา
// รายงานแสดงล ําดับ รหัสสินคา ชื่อ จํ านวนเหลือ จํ านวนสั่งเพิ่ม และจุดสั่งซื้อ
PROCEDURE S0403
   USE PRO INDEX PRO1
   REINDEX
   COUNT TO CNTREORDER FOR PROREST < REORDER
   @ 5,5 SAY "จ ํานวนสินคาที่ถึงจุดสั่งซื้อ = " + STR(CNTREORDER)
   IF CNTREORDER > 0
      CLS ; GO TOP ; I = 1
      DO WHILE !EOF()
         IF PROREST < REORDER
            @ ROW()+1,2 SAY I
            @ ROW(),15 SAY PRO
            @ ROW(),25 SAY PRONAME
            @ ROW(),35 SAY PROREST
            @ ROW(),45 SAY PROONORD
            @ ROW(),55 SAY REORDER
            I++
         ENDIF
         SKIP
      ENDDO
   ELSE
      @ 6,5 SAY "ยงัไมมีสินคาชนิดใดถึงจุดสั่งซื้อ"
   ENDIF
   INKEY(3)
   CLOSE ALL
RETURN

PDF created with FinePrint pdfFactory Pro trial version http://www.pdffactory.com

http://www.pdffactory.com


334                                           การเขียนโปรแกรมส ําหรับฐานขอมูลดวย Clipper                                                         

: ตัวอยางที่ 6.27
&   ผลลัพธ
// พมิพรายการสินคาที่ ถึงจุดสั่งซื้อตามผูผลิต และยังไมไดสั่งเพิ่ม
// รายงานแสดงล ําดับ รหัสสินคา ชื่อ จํ านวนเหลือ จํ านวนสั่งเพิ่ม จุดสั่งซื้อ
// รายงานหยุดใหดูหนาละ 20 รายการ
PROCEDURE S0404
   _SUP = 0
   @ 5,5 SAY "รหัสผูผลิต : " GET _SUP PICT "999"
   READ
   USE PRO INDEX PRO1
   REINDEX
   CLS  ;  I := 0 ; J := 0
   DO WHILE !EOF()
      IF REORDER > PROREST .AND. SUP = _SUP .AND. PROONORD = 0
         I++ ; J++
         @ J,2  SAY I
         @ J,15 SAY PRO
         @ J,25 SAY PRONAME
         @ J,35 SAY PROREST
         @ J,45 SAY PROONORD
         @ J,55 SAY REORDER
         IF MOD(I,20) = 0
            INKEY(3)
            J := 0
            CLS
         ENDIF
      ENDIF
      SKIP
   ENDDO
   INKEY(3)
   CLOSE ALL
RETURN

: ตัวอยางที่ 6.28
&   ผลลัพธ
// พมิพรายการสินคาที่เหลือไมถึง 10 เปอรเซนตจะถึงจุดสั่งซื้อ
// รายงานแสดงล ําดับ รหัสสินคา ชื่อ จํ านวนเหลือ จํ านวนสั่งเพิ่ม จุดสั่งซื้อ
// รายงานหยุดใหดูหนาละ 20 รายการ
// สามารถเลือกปลายทางของผลลัพธได
PROCEDURE S0405
   USE PRO INDEX PRO1
   REINDEX
   COUNT TO CNTREORDER FOR PROREST < REORDER * 1.1
   @ 5,5 SAY "จ ํานวนสินคาใกล ถึงจุดสั่งซื้อ = " + STR(CNTREORDER)
   IF CNTREORDER > 0
      INKEY(1) ; CLS
      GASKPRT = ASKPRT()
      IF GASKPRT = 2 .AND. ISPRINTER() // ฟงกชันตรวจการมีอยูของเครื่องพิมพ
         SET DEVICE TO PRINTER
      ENDIF
      SET FILTER TO PROREST < REORDER  * 1.1
      GO TOP
      J = 3
      FOR I = 1 TO CNTREORDER
         @ J,2 SAY STR(I) + " | "
         @ J,15 SAY STR(PRO) + " | "
         @ J,25 SAY LEFT(PRONAME,8) + " | "
         @ J,35 SAY STR(PROREST) + " | "
         @ J,45 SAY STR(PROONORD) + " | "
         @ J,55 SAY STR(REORDER) + " | "
         SKIP
         J++
         IF J > 22
            INKEY(0) ; CLS
            J = 3
            IF GASKPRT = 2  // ส ําหรับเครื่องพิมพ
               EJECT

            ENDIF
         ENDIF
      NEXT
      IF GASKPRT = 2
         EJECT ; SET DEVICE TO SCREEN
      ENDIF
   ELSE
      @ 6,5 SAY "ยงัไมมีสินคาชนิดใดใกล ถึงจุดสั่งซื้อ"
   ENDIF
   INKEY(3)
   CLOSE ALL
RETURN
: ตัวอยางที่ 6.29
&   ผลลัพธ
// พมิพรายการสินคาที่ก ําลังสั่งซื้อเพิ่มปจจุบัน โดยเรียงตามชื่อสินคา
// รายงานเลือกเรคอรดที ่จํ านวนสั่งเพิ่มมากกวา 0
// รายงานแสดงล ําดับ รหัสสินคา จํ านวนเหลือ จํ านวนสั่งเพิ่ม จุดสั่งซื้อ
PROCEDURE S0406
   USE PRO
   SORT ON PRONAME TO TEMPF
   USE TEMPF
   COPY TO PRO
   USE PRO
   I = 1
   DO WHILE !EOF()
      IF PROONORD > 0
         ? I,PRO,PROREST,PROONORD,REORDER
         I++
      ENDIF
      SKIP
   ENDDO
   INKEY(3)
   CLOSE
RETURN

: ตัวอยางที่ 6.30
&   ผลลัพธ
// พมิพรายการสินคาที่ขายดีจากมากไปนอย ตามชวงวันที่
// โดยระบุชวงวันเริ่มตน และวันสิ้นสุด ของวันสั่งสินคา
// รายงานแสดงเลขที่ใบสั่ง ปริมาณขาย และวันที่สั่ง
// โดยขอมูลถูกจัดเรียงตามปริมาณขาย จากมากไปนอย
PROCEDURE S0407
   USE ORDERD
   SORT ON OQUANT/D TO TEMPF
   USE TEMPF
   COPY TO ORDERD
   SELE 1 ; USE PRO INDEX PRO1
   SELE 2 ; USE ORDER
            INDEX ON ORDER TO ORDER1
   SELE 3 ; USE ORDERD
            SET RELATION TO ORDER INTO ORDER,PRO INTO PRO
   _DATEORD1 := _DATEORD2 := CTOD('')
   @ 5,5 SAY "วนัที่สั่งสินคาเริ่มตน = " GET _DATEORD1
   @ 6,5 SAY "วนัที่สั่งสินคาสิ้นสุด = " GET _DATEORD2
   READ
   SET FILTER TO ORDER->DATEORD >= _DATEORD1 .AND. ;
                 ORDER->DATEORD <= _DATEORD2
   GO TOP
   _PROO = 0
   DO WHILE !EOF()
      IF PRO != _PROO
         ? "*  ",PRO,LEFT(PRO->PRONAME,10)
         _PROO = PRO
      ENDIF
      ? ORDER,OQUANT,ORDER->DATEORD
      SKIP
   ENDDO
   INKEY(3)   ;   CLOSE ALL
RETURN

6.4.6 โปรแกรมตาง ๆ ใน SALE5.PRG
โปรแกรมใน SALE5.PRG นี้มีโปรแกรมยอยภายในทั้งหมด 7 โปรแกรม  โดยรวมโปรแกรมที่เกี่ยวกับระบบ

อืน่ เชน โปรแกรมพิมพสรุปคาขนสงใหบริษัทขนสง  โปรแกรมพิมพจดหมายขอโทษที่สงสินคาลาชากวาวันที่ลูกคา
ตองการสนิคา  โปรแกรมพิมพขอมูลจากแฟมอยางงาย ๆ   โปรแกรมพิมพรายชื่อลูกคาที่พนักงานรับผิดชอบ

ระบบอืน่ เปนระบบที่รวมงานตาง ๆ ที่ไมเขากับระบบยอยอื่น ๆ ซึ่งเปนงานที่หลายหลาย  หากตองการรวม
งานในระบบนีเ้ขากับระบบรายงานก็ยอมท ําได เพราะโปรแกรมเหลานี้เนนการจัดทํ ารายงาน แตผูเขียนไมรวมเขาไป
เพราะงานในระบบรายงานมีมาก  จึงแยกโปรแกรมบางอยางมาสรางระบบใหมจนไดระบบอื่นขึ้นมา

ในหวัขอนีจ้ะแสดงตัวอยางโปรแกรมตาง ๆ ที่มีในโปรแกรม SALE5.PRG ทั้งหมด 7 โปรแกรม โดยเริ่มจาก
ตวัอยางที ่6.31 ถึง ตัวอยางที่ 6.37 ( R0501 - R0507 )
: ตัวอยางที่ 6.31
&   ผลลัพธ
// พมิพสรุปคาขนสงใหบริษัทขนสง โดยระบุเดือน
// รายงานแสดงรหัสบริษัทขนสง ชื่อบริษัทขนสง และจ ํานวนเงินจากการขนสง
PROCEDURE S0501
   _MONTH = 0
   @ 5,5 SAY "เดอืนที่ตองการ : " GET _MONTH PICT "99"
   READ
   SELE 1 ; USE SHIPPER
            INDEX ON SHIPPER TO SHIPPER1
   SELE 2 ; USE ORDER
            INDEX ON SHIPPER TO ORDER2
            SET RELATION TO SHIPPER INTO SHIPPER
   SET FILTER TO MONTH(DATEORD) = _MONTH
   GO TOP
   _SHIPPERO = 0   ;   _TFREIGHT = 0
   DO WHILE !EOF()
      IF SHIPPER != _SHIPPERO
         ? "========================="
         ? SHIPPER,SHIPPER->SHIPNAME
         _SHIPPERO = SHIPPER
      ENDIF
      ? ORDER,DATEORD,FREIGHT
      _TFREIGHT = _TFREIGHT + FREIGHT

      SKIP
      IF EOF() .OR. _SHIPPERO != SHIPPER
         ? "*" , _TFREIGHT
         _TFREIGHT = 0
      ENDIF
   ENDDO
   INKEY(3)
   CLOSE ALL
RETURN
: ตัวอยางที่ 6.32
&   ผลลัพธ
// พมิพสถติกิารใชบริการบริษัทขนสงของลูกคาแตละราย โดยระบุชวงวันที่
// แยกกลุมขอมูลเปน 2 ระดับ คือแยกรหัสลูกคา แลวจึงแยกบริษัทขนสง
// รายงานแสดงจ ํานวนเงินที่ลูกคาแตละราย ใชบริการแตละบริษัทเทาใด
// สดุทายของลกูคาแตละราย แสดงจํ านวนเงินที่ลูกคาตองจายรวมเทาใด
PROCEDURE S0502
   _DATEORD1 := _DATEORD2 := CTOD('')
   @ 5,5 SAY "วนัที่สั่งสินคาเริ่มตน = " GET _DATEORD1
   @ 6,5 SAY "วนัที่สั่งสินคาสิ้นสุด = " GET _DATEORD2
   READ
   USE ORDER
   INDEX ON STR(CUST)+STR(SHIPPER) TO ORDER3
   SET FILTER TO ORDER->DATEORD >= _DATEORD1 .AND. ;
                 ORDER->DATEORD <= _DATEORD2
   GO TOP
   _CUSTO := _SHIPPERO := 0
   _TFREIGHTC := _TFREIGHTS := 0
   DO WHILE !EOF()
      IF _CUSTO != CUST
         ? "***",CUST
         _CUSTO = CUST
         _SHIPPERO = 0
      ENDIF
      IF _SHIPPERO != SHIPPER

         ? "  *",SHIPPER
         _SHIPPERO = SHIPPER
      ENDIF
      _TFREIGHTC = _TFREIGHTC + FREIGHT
      _TFREIGHTS = _TFREIGHTS + FREIGHT
      SKIP
      IF EOF() .OR. _SHIPPERO != SHIPPER
         ?? _TFREIGHTS
         _TFREIGHTS := 0
      ENDIF
      IF EOF() .OR. _CUSTO != CUST
         ? "  TOTAL = ",_TFREIGHTC
         _TFREIGHTC := 0
      ENDIF
   ENDDO
   INKEY(3)
   CLOSE ALL
RETURN
: ตัวอยางที่ 6.33
&   ผลลัพธ
// พมิพจดหมายขอโทษที่สงสินคาไมทันก ําหนด โดยระบุเลขที่ใบสงสินคา
// สามารถเลือกปลายทางของผลลัพธได
PROCEDURE S0503
   _ORDER = 0
   @ 4,5 SAY "ระบุเลขที่ใบสงสินคา : " GET _ORDER PICT "99999"
   READ
   IF LASTKEY() != 27
      GASKPRT = ASKPRT()
      IF GASKPRT = 2
         SET DEVICE TO PRINTER
      ENDIF
      @ 5,5 SAY "เรียน ทานผูมีอุปการะคุณ"
      @ 6,5 SAY "ตามที่ทานไดสั่งสินคากับบริษัทของเรา"
      @ 7,5 SAY "โดยมรีายการตามใบสั่งสินคาเลขที ่"+STR(_ORDER)

      @ 8,5 SAY "แลวทางบริษัทของเราไดสงสินคาใหทานชากวากํ าหนด"
      @ 9,5 SAY "กระผมในฐานะประธานกรรมการของบริษัท"
      @ 10,5 SAY "ใครขออภัยเปนอยางสูง และไดช ําระคาปรับมาพรอมกันนี้"
      @ 11,5 SAY "ทายนี ้กระผมขอใหคํ ามั่นวาเราจะปรับปรุงระบบจัดสง"
      @ 12,5 SAY "ใหเรว็ทันก ําหนดเวลาที่ทานตองการทุกครั้งไป"
      @ 13,5 SAY "   จงึเรียนมาเพื่อกราบขออภัยมา ณ ที่นี"้
      @ 14,5 SAY "                      ประธานบริษัทขายสง จํ ากัด"
      IF GASKPRT = 2
         EJECT
         SET DEVICE TO SCREEN
      ENDIF
      INKEY(3)
   ENDIF
RETURN
: ตัวอยางที่ 6.34
&   ผลลัพธ
// พมิพชื่อลูกคาที่พนักงานแตละคนรับผิดชอบ โดยระบุชวงวันที่
// รายงานแยกกลุมขอมูลตามรหัสพนักงาน
// รายงานแสดงรหัสและชื่อพนักงาน รหัสลูกคา ชื่อลูกคา เลขที่ใบสั่ง วันที่สั่ง
PROCEDURE S0504
   _DATEORD1 := _DATEORD2 := CTOD('')
   @ 5,5 SAY "วนัที่สั่งสินคาเริ่มตน = " GET _DATEORD1
   @ 6,5 SAY "วนัที่สั่งสินคาสิ้นสุด = " GET _DATEORD2
   READ
   IF LASTKEY() != 27
      SELE 1 ; USE EMPL INDEX EMPL1
      SELE 2 ; USE CUST
               INDEX ON CUST TO CUST1
      SELE 3 ; USE ORDER
               INDEX ON EMPL TO ORDER4
               SET RELATION TO CUST INTO CUST,EMPL INTO EMPL
               SET FILTER TO DATEORD >= _DATEORD1 .AND. ;
                             DATEORD <= _DATEORD2
      GO TOP

PDF created with FinePrint pdfFactory Pro trial version http://www.pdffactory.com

http://www.pdffactory.com


342                                           การเขียนโปรแกรมส ําหรับฐานขอมูลดวย Clipper                                                         

      _EMPLO = 0
      DO WHILE !EOF()
         IF EMPL != _EMPLO
            ? "*",EMPL,EMPL->EMPLNAME
            _EMPLO = EMPL
         ENDIF
         ? "-",CUST,CUST->CUSTNAME,ORDER,DATEORD
         SKIP
      ENDDO
      INKEY(3)
      CLOSE ALL
   ENDIF
RETURN
: ตัวอยางที่ 6.35
&   ผลลัพธ
// พมิพเลขที่ใบสั่งสินคาที่สงไมทันก ําหนด โดยระบุเดือน
// เลอืกเดือนที่ตรงกับวันสั่ง วันขนสง หรือวันที่ลูกคาตองการ
// รายงานแสดงขอมูลทุกฟลดจากแฟมใบสั่งสินคา
PROCEDURE S0505
   _MONTH := 0
   @ 5,5 SAY "เดอืนที่ตองการรายงาน = " GET _MONTH PICT "99"
   READ
   IF LASTKEY() != 27
      USE ORDER
      SET FILTER TO (MONTH(DATEORD) = _MONTH .OR. ;
                    MONTH(DATEREQ) = _MONTH .OR. ;
                    MONTH(DATESHIP) = _MONTH) .AND. ;
                    DATESHIP > DATEREQ
      COUNT TO CNTREC
      IF CNTREC > 0
         GO TOP
         DO WHILE !EOF()
            ? ORDER,CUST,EMPL,DATEORD,DATEREQ,DATESHIP,SHIPPER,FREIGHT
            SKIP

         ENDDO
      ELSE
         @ 6,5 SAY "ไมพบขอมูลที่ถูกสั่งในเดือนนี้เลย"
      ENDIF
      INKEY(3)
      CLOSE ALL
   ENDIF
RETURN
: ตัวอยางที่ 6.36
&   ผลลัพธ
// พมิพเลขที่ใบสั่งสินคาที่มีก ําหนดสงไมเกิน 10 วันนับจากวันนี้
// รายงานแสดงขอมูลทุกฟลด จากแฟมใบสั่งซื้อ
PROCEDURE S0506
   USE ORDER
   SET FILTER TO DATESHIP < DATE() + 10 .AND. ;
                 DATESHIP > DATE()        // วนัที่สงตองยังไมถึง
   COUNT TO  CNTREC
   IF CNTREC > 0
      GO TOP
      DO WHILE !EOF()
         ? ORDER,CUST,EMPL,DATEORD,DATEREQ,DATESHIP,SHIPPER,FREIGHT
         SKIP
      ENDDO
   ELSE
      ? "ไมมรีายการที่ตองจัดสงใน 10 วันขางหนา"
   ENDIF
   INKEY(3)
   CLOSE ALL
RETURN
: ตัวอยางที่ 6.37
&   ผลลัพธ
// พมิพขอมูลจากแฟมตามที่เลือก โดยเลือกแฟมจากตัวเลือก แลวเปดแฟม
// น ําขอมูลในแฟมมาพิมพ โดยใชฟงกชันอานชื่อฟลดอัตโนมัติ
// สามารถเลือกปลายทางของผลลัพธได

PDF created with FinePrint pdfFactory Pro trial version http://www.pdffactory.com

http://www.pdffactory.com


344                                           การเขียนโปรแกรมส ําหรับฐานขอมูลดวย Clipper                                                         

PROCEDURE S0507
   OMENU = ALERTNEW("เลอืกปรับปรุงแฟมตามตองการ",;
          {"แฟมสั่งสินคา","แฟมสั่งสินคาละเอียด","แฟมสินคา",;
           "แฟมพนักงาน","แฟมลูกคา","แฟมบริษัทรับสงของ",;
           "แฟมประเภทสินคา","แฟมผูผลิต","กลับเมนู"})
   IF MOD(OMENU,9) != 0
      GASKPRT = ASKPRT()
      IF !ISPRINTER()
         GASKPRT = 1
         @ 2,5 SAY "บอกใหทราบวา ... เครื่องพิมพปดอยู"
      ENDIF
      MAXLINE = 20
      IF GASKPRT = 2
         SET DEVICE TO PRINTER
         MAXLINE = 35
         ? "รายงานผลทางเครื่องพิมพ ... โปรดรอสักครู"
      ENDIF
      DO CASE
         CASE OMENU = 1 ; USE ORDER
         CASE OMENU = 2 ; USE ORDERD
         CASE OMENU = 3 ; USE PRO
         CASE OMENU = 4 ; USE EMPL
         CASE OMENU = 5 ; USE CUST
         CASE OMENU = 6 ; USE SHIPPER
         CASE OMENU = 7 ; USE CAT
         CASE OMENU = 8 ; USE SUP
      ENDCASE
      ARFLDNAME = ARRAY(AFIELDS())   // สรางอาเรยเทาจํ านวนฟลดของแฟม
      ARFLDTYPE = ARRAY(AFIELDS())   // สรางอาเรยเทาจํ านวนฟลดของแฟม
      ARFLDWIDTH= ARRAY(AFIELDS())   // สรางอาเรยเทาจํ านวนฟลดของแฟม
      AFIELDS(ARFLDNAME,ARFLDTYPE,ARFLDWIDTH)  // เก็บชื่อฟลด
      LN := 1 ; NUM =  1
      DO WHILE !EOF()
         @ LN+2,2 SAY LTRIM(STR(NUM,3)) + " : "
         FOR I = 1 TO AFIELDS()
             IF GASKPRT = 2

                @ LN+2,PCOL() SAY &(ARFLDNAME[I])
                @ LN+2,PCOL() SAY " | "
             ELSE
                @ LN+2,COL() SAY &(ARFLDNAME[I])
                @ LN+2,COL() SAY " | "
             ENDIF
         NEXT
         IF MOD(LN,MAXLINE) = 0 .AND. LN != 0
            IF GASKPRT = 2
               EJECT
            ENDIF
            ? "กดปุมใด ๆ เพื่อพิมพหนาตอไป"
            LN = 0  ; INKEY(3) ; CLS
         ENDIF
         LN++ ; NUM++
         SKIP
      ENDDO
      IF GASKPRT = 2
         EJECT
         SET DEVICE TO SCREEN
      ENDIF
      INKEY(5)
   ENDIF
   CLOSE ALL
RETURN

